

Κεφάλαια 2, 7, 8

Βασικές έννοιες αλγορίθμων - προγραμματισμού

(§2.1) Η έννοια Αλγόριθμος.

Κριτήρια αλγορίθμου:

1. **Είσοδος** (προαιρετική, π.χ. γεννήτρια τυχαίων αριθμών)
2. **Έξοδος**
3. **Καθοριστικότητα** (καμία αμφιβολία για τον τρόπο εκτέλεσης της κάθε εντολής π.χ. διαίρεση με το μηδέν)
4. **Περατότητα** (όταν παραβιάζεται, ο μηχανισμός ονομάζεται "υπολογιστική διαδικασία")
5. **Αποτελεσματικότητα** (κάθε εντολή είναι απλή και εκτελέσιμη)

(§2.2) Σκοπιές μελέτης των αλγορίθμων από την Πληροφορική

- Υλικού ~ αρχιτεκτονική ΗΥ και ταχύτητα του αλγορίθμου (π.χ. RAM)
- Γλωσσών προγραμματισμού ~ ταχύτητα
- Θεωρητική (ποιος ο βέλτιστος αλγόριθμος;)
- Αναλυτική (απαιτήσεις του αλγορίθμου σε υπολογιστικούς πόρους: RAM, CPU, IO)

(§2.3) Τρόποι αναπαράστασης αλγορίθμων:

1. ελεύθερο κείμενο (π.χ. μαγειρική συνταγή) – ο λιγότερο δομημένος τρόπος ☒ αποτελεσματικότητα
2. διαγραμματικές τεχνικές
3. φυσική γλώσσα κατά βήματα (π.χ. manual) ☒ καθοριστικότητα
4. κωδικοποίηση

(τα 1, 2 και 3 απευθύνονται στον άνθρωπο, ενώ το 4 στον ΗΥ)

(§7.1) Στοιχεία της ΓΛΩΣΣΑΣ (αλφάβητο: γράμματα ελληνικά-λατινικά, ψηφία, ειδικά σύμβολα)

(§7.2) Τύποι δεδομένων:

1. Ακέραιος π.χ. 3, -9, 2004
2. Πραγματικός π.χ. 3.14
3. Χαρακτήρας π.χ. "3ο Γενικό Λύκειο Ηρακλείου"
4. Λογικός π.χ. έγγαμος

(§7.3) Σταθερές:

α) σταθερές τιμές: τιμές οποιουδήποτε τύπου, που δεν αλλάζουν κατά την εκτέλεση του προγράμματος.

β) συμβολικές σταθερές: ορισμός, τύποι, δήλωση, παραδείγματα, πλεονεκτήματα (πιο ευανάγνωστο και πιο ευέλικτο σε αλλαγές πρόγραμμα)

Σύνταξη

ΣΤΑΘΕΡΕΣ

Όνομα-1 = σταθερή-τιμή-1

Όνομα-2 = σταθερά-τιμή-2

·

·

·

Όνομα-ν = σταθερά-τιμή-ν

Παραδείγματα

ΣΤΑΘΕΡΕΣ

ΠΙ=3.14159

ΦΠΑ=0.18

ΟΝΟΜΑ='Κώστας'

Διαφοροποίηση προγραμματιστή και χρήστη:

Κόσμος προγραμματιστή	Κόσμος Χρήστη
1. Απαιτείται να γνωρίζει προγραμματισμό (κατασκευή προγραμμάτων) – αποκτάται με σπουδές	1. Απαιτείται να γνωρίζει χειρισμό (χρήση προγραμμάτων) – αποκτάται με σεμινάρια χρήσης
2. Κατασκευάζει / συντηρεί / διορθώνει προγράμματα	2. Χειρίζεται προγράμματα
3. Έχει πρόσβαση στον κώδικα και στη διεπαφή χρήσης του προγράμματος	3. Έχει πρόσβαση ΜΟΝΟ στη διεπαφή χρήσης του προγράμματος (εκτελέσιμο)
4. Τον αφορούν οι έννοιες «ευανάγνωστο πρόγραμμα» και «φιλικότητα προς το χρήστη»	4. Τον ΜΟΝΟ αφορά η έννοια «πρόγραμμα φιλικό προς το χρήστη»

(§7.4) Μεταβλητές: ορισμός, τύποι, δήλωση, παραδείγματα (λειτουργία του ΗΥ: δέσμευση χώρου στη μνήμη με **ΑΠΡΟΣΔΙΟΡΙΣΤΕΣ** τιμές),.

Αντιστοιχίζονται από το μεταγλωττιστή σε συγκεκριμένες θέσεις μνήμης επιτρεπτά ονόματα - κανόνες:

1. όλες οι μεταβλητές να δηλώνονται
2. όχι 2 μεταβλητές με το ίδιο όνομα
3. ονόματα σχετικά με τα δεδομένα → πιο ευανάγνωστο
4. όχι δεσμευμένες λέξεις
5. όχι κενά ενδιάμεσα
6. όχι αριθμοί στην αρχή
7. όχι ειδικά σύμβολα

Σύνταξη

ΜΕΤΑΒΛΗΤΕΣ

τύπος-1: Λίστα-μεταβλητών-1

τύπος-2: Λίστα-μεταβλητών-2

.

.

.

Τύπος-ν: Λίστα-μεταβλητών-ν

Παραδείγματα

ΜΕΤΑΒΛΗΤΕΣ

ΠΡΑΓΜΑΤΙΚΕΣ: Εμβαδόν, Α

ΑΚΕΡΑΙΕΣ: ΤΙΜΗ, Ν

ΧΑΡΑΚΤΗΡΕΣ: Όνομα

ΛΟΓΙΚΕΣ: Έλεγχος

(§7.5) Αριθμητικοί τελεστές: +, -, *, /, ^, ΑΚΕΡΑΙΟΙ τελεστές: div, mod ($x \text{ div } y$, $x \text{ mod } y$: $y \neq 0$ και $x, y > 0$) π.χ.:

$x \text{ mod } 2 = 0 ? \Leftrightarrow$ άρτιος ?

$x \text{ mod } y = 0 ? \Leftrightarrow$ ο x πολλαπλάσιο του y ?

$x \text{ mod } 10 =$ τελευταίο ψηφίο του x

ανάλυση 2/3/4ψηφίου κλπ αριθμού στα ψηφία του

x	y
$x \text{ mod } y$	$x \text{ div } y$

$$x \text{ mod } y \in [0, y-1] \quad x \text{ div } y \in [0, x]$$

$$x \text{ div } y * y + x \text{ mod } y = x$$

x	y	x div y	x mod y
21	7	3	0
18	4	4	2
7	9	0	7

2ψηφιος π.χ. 53: 5=53 div 10, 3=53 mod 10

3ψηφιος π.χ. 538: 5=538 div 100, 3=538 div 10 mod 10, 8=538 mod 10

Έλεγχος αν ο x έχει το πολύ 3 ψηφία

$$x \text{ div } 1000 = 0$$

Τιμή του ψηφίου των χιλιάδων

$$x \text{ div } 1000 \text{ mod } 10$$

Έλεγχος αν ο x έχει τουλάχιστον 3 ψηφία

$$x \text{ div } 100 <> 0$$

Τιμή του ψηφίου των εκατοντάδων

$$x \text{ div } 100 \text{ mod } 10$$

1° δεκαδικό ψηφίο $x \in \mathbb{R}$

$$A_M(x*10) \text{ mod } 10$$

(§7.6) Συναρτήσεις:

HM(X)	Υπολογισμός ημιτόνου
ΣΥΝ(X)	Υπολογισμός συνημιτόνου
ΕΦ(X)	Υπολογισμός εφαπτομένης
T_P(X)	Υπολογισμός τετραγωνικής ρίζας
ΛΟΓ(X)	Υπολογισμός φυσικού λογαρίθμου
E(X)	Υπολογισμός του e^x
A_M(X)	Ακέραιο μέρος του X
A_T(X)	Απόλυτη τιμή του X

δεκαδικό μέρος του X = $X - A_M(X)$

$$T_P(X) = X^{(1/2)}$$

$$A_T(X) = T_P(X^2)$$

Οι τριγωνομετρικές συναρτήσεις: σε μοίρες. Π.χ. $HM(30) = 0.5$

(§7.7) Αριθμητικές εκφράσεις: ορισμός, όλες οι μεταβλητές, πρέπει να έχουν τιμή!, ιεραρχία:

1. Παρενθέσεις, συναρτήσεις
2. Ύψωση σε δύναμη
3. Πολλαπλασιασμός και διαίρεση, div, mod
4. Πρόσθεση και αφαίρεση

π.χ. $2+3*4 = 14$ ενώ $(2+3)*4 = 20$

(§7.8) Εντολή εκχώρησης: λειτουργία, σύνταξη:

<μεταβλητή> ← <έκφραση>

Κανόνες:

- το αριστερό και το δεξιό μέρος πρέπει να είναι του ίδιου τύπου
- στο αριστερό μέρος: μόνο μεταβλητή
- οι μεταβλητές στο δεξιό μέρος πρέπει να ΜΗΝ είναι απροσδιόριστες
- στο αριστερό και το δεξιό μέρος μπορεί να εμφανίζεται η ίδια μεταβλητή.

Μεταβλητές

Ακέραιες: x, y, z, m

Πραγματικές: f

Χαρακτήρες: τίτλος

Λογικές: ισχύει

Εντολή	x	y	z	m	f	τίτλος	ισχύει
	-	-	-	-	-	-	-
$x \leftarrow 5$	5	-	-	-	-	-	-
$x \leftarrow 7$	7	-	-	-	-	-	-
$y \leftarrow x$	7	7	-	-	-	-	-
$x \leftarrow x + 3$	10	7	-	-	-	-	-
$z \leftarrow x + y$	10	7	17	-	-	-	-
$y \leftarrow x + m$	10	?	17	-	-	-	-
$y \leftarrow x^2$	10	100	17	-	-	-	-
$m \leftarrow x + 2 * z$	10	100	17	44	-	-	-
$f \leftarrow (x+z)/2$	10	100	17	44	13.5	-	-
τίτλος $\leftarrow x$	10	100	17	44	13.5	-	-
τίτλος $\leftarrow "3o"$	10	100	17	44	13.5	3o	-
$x+y \leftarrow 9$	10	100	17	44	13.5	3o	-
ισχύει \leftarrow Αληθής	10	100	17	44	13.5	3o	Αληθής
ισχύει $\leftarrow x > y$	10	100	17	44	13.5	3o	Ψευδής

στο αριστερό και το δεξιό μέρος μπορεί να εμφανίζεται η ίδια μεταβλητή

ΛΑΘΟΣ: οι μεταβλητές στο δεξιό μέρος πρέπει να ΜΗΝ είναι απροσδιόριστες

ΛΑΘΟΣ: το αριστερό και το δεξιό μέρος πρέπει να είναι του ίδιου τύπου


ΛΑΘΟΣ: στο αριστερό μέρος: μόνο μεταβλητή

(§7.9) Εντολή εισόδου **ΔΙΑΒΑΣΕ**:

Σύνταξη: Διάβασε <λίστα μεταβλητών>

Λειτουργία: προκαλείται μία «παύση» στην εκτέλεση των εντολών και το πρόγραμμα περιμένει από τον χρήστη την εισαγωγή τόσων τιμών, όσες και οι μεταβλητές της λίστας. Η κάθε εισαγόμενη τιμή, αποθηκεύεται στην αντίστοιχη μεταβλητή.

Παραδείγματα:



Διάβασε x
Διάβασε y
↔
Διάβασε x, y

(άνθρωπος → ΗΥ) για εισαγωγή δεδομένων (στην προστακτική, διότι ο προγραμματιστής-σκηνοθέτης διατάζει τον ΗΥ, ενώ ο χρήστης είναι ο θεατής του έργου).

Διαφορά της εντολής Διάβασε από την εκχώρηση (←)
π.χ. 2βάθμια



(§7.9) Εντολή εξόδου **ΓΡΑΨΕ**:

Σύνταξη: Γράψε <λίστα εκφράσεων>

Λειτουργία: το πρόγραμμα υπολογίζει τις τελικές τιμές των εκφράσεων και τις εμφανίζει στην οθόνη

Παραδείγματα:

Εντολή	Οθόνη
Γράψε 7	7
Γράψε 7^2	49
Γράψε "7^2"	7^2
x ← 5	
Γράψε x	5
y ← 7	
Γράψε x, y	5 7
Γράψε z	?
μο ← (x+y)/2	
Γράψε "μέσος όρος: ", μο	μέσος όρος: 6

οι μεταβλητές πρέπει να ΜΗΝ είναι απροσδιόριστες

(ΗΥ → άνθρωπος) για εξαγωγή αποτελεσμάτων



(§7.10) Δομή προγράμματος / Αλγορίθμου:

Πρόγραμμα <όνομα πρ/τος>
[Δήλωση σταθερών]
<Δήλωση μεταβλητών>
Αρχή

Σώμα εντολών:

Διάβασε ...

Εκχωρήσεις ...

Γράψε ...

Τέλος_Προγράμματος

Αλγόριθμος <όνομα αλγορίθμου>
[Αρχή]

Σώμα εντολών:

Διάβασε ... ή Δεδομένα // ... //

Εκχωρήσεις ...

Γράψε ... ή Εμφάνισε ή Αποτελέσματα // ... //

Τέλος

(§7.10) Δομή προγράμματος / Αλγόριθμος: (π.χ. Πρόγραμμα που διαβάζει τις διαστάσεις ενός τραpezίου και εμφανίζει το εμβαδόν του)


Μη φιλικά προς τον χρήστη	Φιλικά προς τον χρήστη
<p>Πρόγραμμα Τραπεζίο Μεταβλητές Πραγματικές: β, Β, υ, Ε Αρχή Διάβασε β, Β, υ $E \leftarrow (\beta+B)*\upsilon/2$ Γράψε Ε Τέλος_Προγράμματος</p>	<p>Πρόγραμμα Τραπεζίο Μεταβλητές Πραγματικές: β, Β, υ, Ε Αρχή Γράψε "Δώσε διαστάσεις τραpezίου" Διάβασε β, Β, υ $E \leftarrow (\beta+B)*\upsilon/2$ Γράψε "Εμβαδό: ", Ε Τέλος_Προγράμματος</p>




(§2.4) Διάγραμμα ροής (Δ.Ρ.)

ΑΠΑΙΤΕΙΤΑΙ η γνώση μετατροπής μεταξύ των 3 εννοιών:


Κώδικας \leftrightarrow Δ.Ρ. \leftrightarrow Εκφώνηση



π.χ. $MO \leftarrow (x+y+z)/3$



π.χ. $ax + b = 0$



Σχόλια

Ορισμός: οτιδήποτε μπαίνει μετά από το ειδικό σύμβολο !

Χρησιμότητα: δεν επηρεάζουν τη λειτουργία του προγράμματος,
αλλά το κάνουν πιο ευανάγνωστο

π.χ. μισθός ← $1.05 * \text{μισθός}$! προσαύξηση μισθού κατά 5%

Ο ειδικός χαρακτήρας &: συνέχεια κώδικα και στην επόμενη γραμμή.

Αλγοριθμικές Δομές:

1. Ακολουθία
2. Επιλογή
3. Επανάληψη

(§2.4.1) Παραδείγματα δομής ακολουθίας (και Δ.Ρ.)

Πραγματικοί $a, \beta \rightarrow$ απόσταση των a και $\beta = A_T(a-\beta)$ π.χ. απόκλιση προφορικού και γραπτού βαθμού

```
Διάβασε π, γ  
απ ← A_T(π - γ)  
Γράψε απ
```

Θερμοκρασία C \rightarrow Θερμοκρασία F: $(F-32)/9 = C/5$

```
Διάβασε C  
F ← 9 * C / 5 + 32  
Γράψε F
```

Κεφάλαιο, % επιτόκιο \rightarrow τελικό κεφάλαιο μετά από τρία έτη

```
Διάβασε K, E  
K ← K + E/100*K  
K ← K + E/100*K  
K ← K + E/100*K  
Γράψε K
```

(§2.4.1) Παραδείγματα δομής ακολουθίας (και Δ.Ρ.)

R κύκλου → διάμετρος, περιφέρεια, εμβαδό, όγκος σφαίρας με αυτήν την ακτίνα (πR^3)

Διάβασε R

$$\Delta \leftarrow 2 * R$$

$$\Pi \leftarrow 2 * 3.14 * R$$

$$E \leftarrow 3.14 * R^2$$

$$V \leftarrow 3.14 * R^3$$

Γράψε Δ, Π, E, V

a, β ορθογώνιου παρ/γμου → περίμετρος, εμβαδό, διαγώνιος

Διάβασε a, β

$$\Pi \leftarrow 2 * (a + \beta)$$

$$E \leftarrow a * \beta$$

$$\Delta \leftarrow T_P(a^2 + \beta^2)$$

Γράψε Π, E, Δ

S : εμβαδό κύκλου → εμβαδό του περιγεγραμμένου 4γώνου

Διάβασε S

$$R \leftarrow T_P(S / 3.14)$$

$$a \leftarrow 2 * R$$

$$E \leftarrow a^2$$

Γράψε E

(§2.4.1) Παραδείγματα δομής ακολουθίας (και Δ.Ρ.)

$(x_1, y_1), (x_2, y_2) \rightarrow$ απόσταση των 2 σημείων

```
Διάβασε  $x_1, y_1, x_2, y_2$   
 $d \leftarrow T\_P((x_1-x_2)^2 + (y_1-y_2)^2)$   
Γράψε  $d$ 
```

Άθροισμα ψηφίων 3ψήφιου θετικού ακεραίου

```
Διάβασε  $x$   
 $x_1 \leftarrow x \text{ div } 100$   
 $x_2 \leftarrow x \text{ div } 10 \text{ mod } 10$   
 $x_3 \leftarrow x \text{ mod } 10$   
 $S \leftarrow x_1 + x_2 + x_3$   
Γράψε  $S$ 
```

Ένα παιχνίδι παίζεται κυκλικά από 5 παίκτες. Σε κάθε γύρο παίζει 1 παίκτης. Με δεδομένο τον γύρο (≥ 1) του παιχνιδιού ποιος παίκτης (1-5) έπαιξε;

```
παίκτης  $\leftarrow (\text{γύρος}-1) \text{ mod } 5 + 1$ 
```

(§2.4.1) Παραδείγματα δομής ακολουθίας (και Δ.Ρ.)

$t(\text{sec}) \rightarrow \text{days, hours, min, sec}$ (1 day = 86400 sec)

```
Διάβασε t
d ← t div 86400
t ← t mod 86400
h ← t div 3600
t ← t mod 3600
m ← t div 60
s ← t mod 60
Γράψε d, h, m, s
```

Ακέραιος $x \rightarrow x-x'$ (π.χ. $785 \rightarrow 785 - 587$)

```
Διάβασε x
x1 ← x div 100
x2 ← x div 10 mod 10
x3 ← x mod 10
sx ← x3 * 100 + x2 * 10 + x1
d ← x - sx
Γράψε d
```

Αλυσίδα Super Market προσφέρει δωροεπιταγή 3€ για κάθε 100 πόντους αγορών. Εισαγωγή των πόντων ενός πελάτη και εμφάνιση: πόσες δωροεπιταγές δικαιούται και πόσοι πόντοι αγορών του περισσεύουν.

```
Διάβασε π
Γράψε π div 100, ' δωροεπιταγές ', π mod 100, ' πόντοι περισσεύουν'
```

(§2.4.1) Παραδείγματα δομής ακολουθίας (και Δ.Ρ.)

"Επιμερισμός ποσότητας" π.χ. να γραφεί πρόγραμμα που διαβάζει το συνολικό ποσό επιχορήγησης τριών δήμων και τους πληθυσμούς τους. Εάν το ποσό διανέμεται αναλογικά με βάση τον πληθυσμό του κάθε δήμου, να εμφανίζεται το ποσό που θα λάβει ο κάθε δήμος

```
ΓΡΑΨΕ 'Δώσε ποσό επιχορήγησης'  
ΔΙΑΒΑΣΕ poso  
ΓΡΑΨΕ 'Δώσε τους πληθυσμούς των τριών δήμων'  
ΔΙΑΒΑΣΕ p1 ,p2, p3  
s <-- p1+p2+p3  
ΓΡΑΨΕ 'Ποσά επιχορήγησης των δήμων:'  
ΓΡΑΨΕ p1/s*poso, p2/s*poso, p3/s*poso, '€'
```

Συμπληρώστε το κενό ώστε να εμφανισθεί η τιμή 24

```
x ← 4  
y ← x * 3  
x ← x - ...  
y ← x * y  
Γράψε y
```

x	y
4	-
4	12
4-ω	12
4-ω	12(4-ω)

$12(4-\omega) = 24 \Rightarrow \omega = 2$

Γράψτε την ισοδύναμη έκφραση: $\frac{7}{(y-9)^{\frac{3}{4 \cdot y}}}$

$$7/(y-9)^{(3/(4*y))}$$

(§2.4.1) Παραδείγματα δομής ακολουθίας (και Δ.Ρ.)

Ποσοστά

(αρχική τιμή, %μεταβολή_ποσ) → τελική τιμή
 $\text{τελική_τιμή} \leftarrow \text{αρχική_τιμή} + \text{αρχική_τιμή} * \text{μεταβολή_ποσ}/100$

(αρχική τιμή, %μεταβολή_ποσ) → μεταβολή (τιμή)
 $\text{μεταβολή_τιμή} \leftarrow \text{αρχική_τιμή} * \text{μεταβολή_ποσ}/100$

(τελική τιμή, %μεταβολή_ποσ) → αρχική τιμή
 $\text{αρχική_τιμή} \leftarrow \text{τελική_τιμή} / (1 + \text{μεταβολή_ποσ}/100)$

(αρχική τιμή, τελική τιμή) → %μεταβολή_ποσ
 $\text{μεταβολή_ποσ} \leftarrow (\text{τελική_τιμή} - \text{αρχική_τιμή}) / \text{αρχική_τιμή} * 100$

(αρχική τιμή, μεταβολή) → %μεταβολή_ποσ
 $\text{μεταβολή_ποσ} \leftarrow \text{μεταβολή} / \text{αρχική_τιμή} * 100$

(σύνολο, υποσύνολο) → %ποσοστό
 $\text{ποσοστό} \leftarrow \text{υποσύνολο} / \text{σύνολο} * 100$

(σύνολο, %ποσοστό) → υποσύνολο
 $\text{υποσύνολο} \leftarrow \text{σύνολο} * \text{ποσοστό}/100$

(υποσύνολο, %ποσοστό) → σύνολο
 $\text{σύνολο} \leftarrow \text{υποσύνολο} / \text{ποσοστό} * 100$

(§2.4.1) Παραδείγματα δομής ακολουθίας (και Δ.Ρ.)

Αντιμετάθεση τιμών δύο μεταβλητών x και y

$x \leftarrow y$
 $y \leftarrow x$


	x	y
	5	7
$x \leftarrow y$	7	7
$y \leftarrow x$	7	7

ΛΑΘΟΣ

$tmp \leftarrow x$
 $x \leftarrow y$
 $y \leftarrow tmp$

	x	y	tmp
	5	7	-
$tmp \leftarrow x$	5	7	5
$x \leftarrow y$	7	7	5
$y \leftarrow tmp$	7	5	5

ΣΩΣΤΟ




$x \leftarrow x + y$
 $y \leftarrow x - y$
 $x \leftarrow x - y$

ΕΝΑΛΛΑΚΤΙΚΑ
(μόνο για x, y :
αριθμητικές)

$tmp \leftarrow x + y$
 $y \leftarrow tmp - y$
 $x \leftarrow tmp - y$

(§2.4.1) Παραδείγματα δομής ακολουθίας (και Δ.Ρ.)

Κυκλική αντιμετάθεση τιμών τριών μεταβλητών x, y, z



```
tmp ← x
x ← y
y ← z
z ← tmp
```

x	y	z	tmp
5	7	9	-
5	7	9	5
7	7	9	5
7	9	9	5
7	9	5	5

Εισαγωγή 4ψήφιου ακέραιου x και 1ψήφιου $y \rightarrow$ 5ψήφιος με τον y : μεσαίο ψηφίο

```
Διάβασε  $x, y$ 
 $x1 \leftarrow x \text{ div } 100$ 
 $x2 \leftarrow x \text{ mod } 100$ 
 $z \leftarrow x1 * 1000 + y * 100 + x2$ 
Γράψε  $z$ 
```

(§2.4.2-4 - §8.1) Δομή επιλογής

α. Λογική έκφραση: <έκφραση1> συγκριτικός τελεστής <έκφραση 2>

β. Τελεστές σύγκρισης

Συγκριτικοί τελεστές		
Τελεστής	Ελεγχόμενη σχέση	Παράδειγμα
=	Ισότητα	Αριθμός=0
<>	Ανισότητα	Όνομα1 <> 'Κώστας'
>	Μεγαλύτερο από	Τιμή>10000
>=	Μεγαλύτερο ή ίσο	$X+Y \geq (A+B)/\Gamma$
<	Μικρότερο από	$B^2-4*A*\Gamma < 0$
<=	Μικρότερο ή ίσο	Βάρος <= 500

(Παρατηρήσεις:

α) η σύγκριση των λεκτικών τιμών γίνεται με αλφαβητική σειρά.
π.χ. "ΑΛΕΞΑΝΔΡΟΣ" > "ΑΛΕΞΑΝΔΡΑ"

β) η σύγκριση λογικών δεδομένων έχει έννοια μόνο στην περίπτωση του ίσου (=) και του διάφορου (<>))

γ) η μεταβλητή τύπου χαρακτήρων λέξη ξεκινάει με Κ: (λέξη >= "Κ" ΚΑΙ λέξη < "Λ")

(§2.4.2-4 - §8.1) Δομή επιλογής

γ. Σύνθετη λογική έκφραση:

<λογική έκφραση 1> λογικός τελεστής <λογική έκφραση 2>

δ. Λογικοί τελεστές

i. Τελεστής ΚΑΙ (σύζευξη)

Σύνταξη: <λογική έκφραση 1> ΚΑΙ <λογική έκφραση 2>

Λειτουργία: ισούται με Αληθής όταν όλες οι εκφράσεις που συνδέει, είναι Αληθείς

Παράδειγμα: $x > 0$ ΚΑΙ $x \leq 12$ ($\Leftrightarrow x \in (0, 12]$)

ii. Τελεστής Η (διάζευξη)

Σύνταξη: <λογική έκφραση 1> Η <λογική έκφραση 2>

Λειτουργία: ισούται με Αληθής όταν τουλάχιστον μία από τις εκφράσεις που συνδέει, είναι Αληθής

Παράδειγμα: $x \leq 0$ Η $x > 12$ ($\Leftrightarrow x \in (-\infty, 0] \cup (12, +\infty)$)

(§2.4.2-4 - §8.1) Δομή επιλογής

iii. Τελεστής ΟΧΙ (άρνηση)

Σύνταξη: ΟΧΙ <λογική έκφραση>

Λειτουργία: Ισούται με την αντίθετη τιμή της έκφρασης

Παραδείγματα:

1. ΟΧΙ ($x = 7$) ($\Leftrightarrow x \neq 7$)
2. ΟΧΙ ($y > 11$) ($\Leftrightarrow y \leq 11$)
3. ΟΧΙ ($z > -3$ ΚΑΙ $z \leq 18$) ($\Leftrightarrow z \leq -3$ Η $z > 18$)
4. ΟΧΙ ($\omega \leq 0$ Η $\omega > 21$) ($\Leftrightarrow \omega > 0$ ΚΑΙ $\omega \leq 21$)

Πίνακας Αληθείας

Πρόταση Α	Πρόταση Β	Α ή Β	Α και Β	όχι Α
Αληθής	Αληθής	Αληθής	Αληθής	Ψευδής
Αληθής	Ψευδής	Αληθής	Ψευδής	Ψευδής
Ψευδής	Αληθής	Αληθής	Ψευδής	Αληθής
Ψευδής	Ψευδής	Ψευδής	Ψευδής	Αληθής


(§2.4.2-4 - §8.1) Δομή επιλογής

ε. Ιεραρχία λογικών τελεστών:

1. Παρενθέσεις
2. ΟΧΙ
3. ΚΑΙ
4. Η

στ. Ιεραρχία τελεστών:

1. Αριθμητικοί (+, -, κλπ.)
2. Συγκριτικοί (>, =, κλπ.)
3. Λογικοί (ΟΧΙ, ΚΑΙ, Η)



(§2.4.2-4 - §8.1) Δομή επιλογής

ζ. Απλή Δομή Αν – ΤέλοςΑν

Σύνταξη:

Αν <συνθήκη> τότε

<εντολές>

ΤέλοςΑν

ή Αν <συνθήκη> τότε <εντολή>

π.χ. $a \rightarrow |a|$



Διάβασε a

Αν $(a < 0)$ τότε

$a \leftarrow (-1) * a$

ΤέλοςΑν

Γράψε a



(§2.4.2-4 - §8.1) Δομή επιλογής

η. Σύνθετη Δομή Αν - Αλλιώς - ΤέλοςΑν

Σύνταξη:


Αν <συνθήκη> τότε

<εντολές1>

Αλλιώς

<εντολές2>

ΤέλοςΑν



π.χ. θετικός ακέραιος $a \rightarrow$ άρτιος / περιττός

Διάβασε a


Αν $(a \bmod 2 = 0)$ τότε

Γράψε "άρτιος"

Αλλιώς

Γράψε "περιττός"

ΤέλοςΑν




(§2.4.2-4 - §8.1) Δομή επιλογής

θ. Εμφωλευμένες Αν

ΚΑΝΟΝΑΣ ΕΜΦΩΛΕΥΣΗΣ: Κάθε Αλλιώς και Τέλος Αν αντιστοιχεί στο πλησιέστερο ανοικτό Αν

Να διαβάζονται δύο αριθμοί που αντιστοιχούν στο ύψος και βάρος ενός άνδρα. Να εκτυπώνεται ότι ο άνδρας είναι “ελαφρύς”, αν το βάρος του είναι κάτω από 80 κιλά, ή να εκτυπώνεται “βαρύς” στην αντίθετη περίπτωση. Επίσης να εκτυπώνεται “κοντός” αν το ύψος του είναι κάτω από 1.70, αλλιώς να εκτυπώνεται “ψηλός”.


```
Αλγόριθμος Παράδειγμα_6
Διάβασε βάρος, ύψος
Αν βάρος < 80 τότε
  Αν ύψος < 1.70 τότε
    εκτύπωσε 'Ελαφρύς, κοντός'
  αλλιώς
    εκτύπωσε 'ελαφρύς, ψηλός'
Τέλος_αν
αλλιώς
  Αν ύψος < 1.70 τότε
    εκτύπωσε 'Βαρύς, κοντός'
  αλλιώς
    εκτύπωσε 'βαρύς, ψηλός'
Τέλος_αν
Τέλος_αν
Τέλος Παράδειγμα_5
```



(§2.4.2-4 - §8.1) Δομή επιλογής

θ. Εμφωλευμένες Αν

Εισαγωγή των συντεταγμένων (x,y) ενός σημείου $(x,y \neq 0)$ και εμφάνιση του τεταρτημορίου (1-4) στο οποίο βρίσκεται:



```
ΠΡΟΓΡΑΜΜΑ tetarthmoria
ΜΕΤΑΒΛΗΤΕΣ
  ΠΡΑΓΜΑΤΙΚΕΣ: x, y
ΑΡΧΗ
  ΔΙΑΒΑΣΕ x, y
  ΑΝ x >0 ΤΟΤΕ
    ΑΝ y >0 ΤΟΤΕ
      ΓΡΑΨΕ 1
    ΑΛΛΙΩΣ
      ΓΡΑΨΕ 2
    ΤΕΛΟΣ_ΑΝ
  ΑΛΛΙΩΣ
    ΑΝ y >0 ΤΟΤΕ
      ΓΡΑΨΕ 4
    ΑΛΛΙΩΣ
      ΓΡΑΨΕ 3
    ΤΕΛΟΣ_ΑΝ
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ
```

ή

```
ΠΡΟΓΡΑΜΜΑ tetarthmoria2
ΜΕΤΑΒΛΗΤΕΣ
  ΠΡΑΓΜΑΤΙΚΕΣ: x, y
ΑΡΧΗ
  ΔΙΑΒΑΣΕ x, y
  ΑΝ x>0 ΚΑΙ y>0 ΤΟΤΕ
    ΓΡΑΨΕ 1
  ΑΛΛΙΩΣ_ΑΝ x>0 ΚΑΙ y<0 ΤΟΤΕ
    ΓΡΑΨΕ 2
  ΑΛΛΙΩΣ_ΑΝ x<0 ΚΑΙ y<0 ΤΟΤΕ
    ΓΡΑΨΕ 3
  ΑΛΛΙΩΣ
    ΓΡΑΨΕ 4
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ
```

ή


```
ΠΡΟΓΡΑΜΜΑ tetarthmoria3
ΜΕΤΑΒΛΗΤΕΣ
  ΠΡΑΓΜΑΤΙΚΕΣ: x, y
ΑΡΧΗ
  ΔΙΑΒΑΣΕ x, y
  ΑΝ x>0 ΚΑΙ y>0 ΤΟΤΕ
    ΓΡΑΨΕ 1
  ΤΕΛΟΣ_ΑΝ
  ΑΝ x>0 ΚΑΙ y<0 ΤΟΤΕ
    ΓΡΑΨΕ 2
  ΤΕΛΟΣ_ΑΝ
  ΑΝ x<0 ΚΑΙ y<0 ΤΟΤΕ
    ΓΡΑΨΕ 3
  ΤΕΛΟΣ_ΑΝ
  ΑΝ x<0 ΚΑΙ y>0 ΤΟΤΕ
    ΓΡΑΨΕ 4
  ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ
```

(§2.4.2-4 - §8.1) Δομή επιλογής

1. Δομή Αν πολλαπλής επιλογής

Αν Σ1 τότε
 εντολές1
Αλλιώς Αν Σ2 τότε
 εντολές2
...
Αλλιώς Αν Σn τότε
 εντολέςn
[Αλλιώς
 εντολέςn+1]
Τέλος Αν

Εκτελούνται οι εντολές που βρίσκονται στο αντίστοιχο τμήμα, όταν η συνθήκη είναι αληθής. Η εκτέλεση του προγράμματος συνεχίζεται με την εντολή που ακολουθεί τη δήλωση ΤΕΛΟΣ_ΑΝ



(§2.4.2-4 - §8.1) Δομή επιλογής

1. Δομή Αν πολλαπλής επιλογής

π.χ. δίν. βαθμός $\beta \in [0-20] \rightarrow$ χαρακτηρισμός :
κακός (≤ 9), μέτριος (9, 14], καλός (14, 16],
πολύ καλός (16, 18], άριστος (18, 20]

Διάβασε β

Αν ($\beta \leq 9$) τότε

Γράψε "κακός"

Αλλιώς Αν ($\beta \leq 14$) τότε

Γράψε "μέτριος"

Αλλιώς Αν ($\beta \leq 16$) τότε

Γράψε "καλός"


Αλλιώς Αν ($\beta \leq 18$) τότε

Γράψε "πολύ καλός"

Αλλιώς

Γράψε "άριστος"

Τέλος Αν



(§2.4.2-4 - §8.1) Δομή επιλογής

κ. Δομή Επίλεξε

Σύνταξη

```
ΕΠΙΛΕΞΕ έκφραση
ΠΕΡΙΠΤΩΣΗ λίστα_τιμών_1
    εντολές_1
ΠΕΡΙΠΤΩΣΗ λίστα_τιμών_2
    εντολές_2
    .....
ΠΕΡΙΠΤΩΣΗ ΑΛΛΙΩΣ
    εντολές_αλλιώς
ΤΕΛΟΣ_ΕΠΙΛΟΓΩΝ
```

<έκφραση> :

είναι μια μεταβλητή, η τιμή της οποίας θα ελεγχθεί με τις τιμές που δίνονται στις ΠΕΡΙΠΤΩΣΕΙΣ και ανάλογα σε ποια ΠΕΡΙΠΤΩΣΗ ανήκει θα εκτελεστούν οι αντίστοιχες εντολές ή η πράξη, που υπολογίζει την τιμή της.

Δηλαδή, η <έκφραση> μπορεί να είναι:

- Μεταβλητή
- Αριθμητική πράξη
- Συγκριτική πράξη

<λίστα_τιμών_N>:

οι τιμές που μπορεί να πάρει μια έκφραση. Οι τιμές αυτές μπορεί να είναι διακριτές τιμές, περιοχή τιμών από...έως ή να υπακούν σε μια συνθήκη.

Να εισαχθεί ένας ακέραιος που αντιστοιχεί σε μια ηλικία και να βρεθεί σε ποια όρια εντάσσεται η δεδομένη ηλικία εμφανίζοντας σχετικό μήνυμα.

Αλγόριθμος Παράδειγμα 5.

Εμφάνισε "Σε ποια ηλικία άρχισες να μαθαίνεις προγραμματισμό;"

Διάβασε age

Επίλεξε age

Περίπτωση < 0

Εμφάνισε "Είπαμε ηλικία ..."

Περίπτωση < 5

Εμφάνισε "Μάλλον τα παραλές !!"

Περίπτωση < 60

Εμφάνισε "Μπράβο"

Περίπτωση < 100

Εμφάνισε "Ποτέ δεν είναι αργά"

Περίπτωση αλλιώς

Εμφάνισε "Κάλλιο αργά παρά ποτέ"

Τέλος_επιλογών

Τέλος Παράδειγμα_5

⊕: συμπαγής δομή

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M1. Εύρεση min/max 3,4... ποσοτήτων (διαδοχικά Αν-ΤέλοςΑν). Για 2 ποσότητες, Αν-Αλλιώς-ΤέλοςΑν

Γενικά: εύρεση π.χ. $\max \{a_1, a_2, \dots, a_n\}$

$\max \leftarrow a_1$

Αν ($a_2 > \max$) τότε

$\max \leftarrow a_2$

ΤέλοςΑν

Αν ($a_3 > \max$) τότε

$\max \leftarrow a_3$

ΤέλοςΑν


...

Αν ($a_n > \max$) τότε

$\max \leftarrow a_n$

ΤέλοςΑν

Γράψε \max



$$\max(x,y) = (x+y)/2 + A_T(x-y)/2$$
$$\min(x,y) = (x+y)/2 - A_T(x-y)/2$$

(x, y: αριθμητικές)

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M2. Εύρεση έννοιας με min/max γνώρισμα. π.χ. 4 μάρκες αυτοκινήτων, 4 τιμές → ακριβότερη μάρκα (χωρίς ισοτιμία)

Διάβασε $\mu_1, \mu_2, \mu_3, \mu_4$

Διάβασε $\tau_1, \tau_2, \tau_3, \tau_4$

$\max \leftarrow \tau_1$

$\mu_{\max} \leftarrow \mu_1$

Αν $(\tau_2 > \max)$ τότε

$\max \leftarrow \tau_2$

$\mu_{\max} \leftarrow \mu_2$

ΤέλοςΑν

Αν $(\tau_3 > \max)$ τότε

$\max \leftarrow \tau_3$

$\mu_{\max} \leftarrow \mu_3$

ΤέλοςΑν


Αν $(\tau_4 > \max)$ τότε

$\max \leftarrow \tau_4$

$\mu_{\max} \leftarrow \mu_4$

ΤέλοςΑν

Γράψε μ_{\max}



Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M3. Ασκήσεις σύνθετων ελέγχων με χρήση των τελεστών: ΚΑΙ, Η, ΟΧΙ
π.χ. Βαθμοί ενός φοιτητή σε 3 μαθήματα → περνάει το δμηνο (έχει σε όλα τουλάχιστον τη βάση (5) ή έχει μέσο όρο άνω του 7)

Διάβασε B1, B2, B3

$MO \leftarrow (B1+B2+B3)/3$


Αν $(B1 \geq 5 \text{ ΚΑΙ } B2 \geq 5 \text{ ΚΑΙ } B3 \geq 5 \text{ Η } MO > 7)$ τότε

Γράψε "Περνάει"

Αλλιώς

Γράψε "Κόβεται"

Τέλος



Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M4. Κλαδικές συναρτήσεις

π.χ. $x \rightarrow f(x)$:

$$f(x) = \begin{cases} x - 1 & \text{για } x < 0 \\ x^2 - 3 \cdot x & \text{για } 0 \leq x \leq 1 \\ \frac{x+1}{x-4} & \text{για } x > 1 \end{cases}$$

Διάβασε x

Αν $(x < 0)$ τότε

$f \leftarrow x-1$

Γράψε f

Αλλιώς Αν $(x \leq 1)$ τότε

$f \leftarrow x^2 - 3 \cdot x$

Γράψε f

Αλλιώς

Αν $(x < 4)$ τότε

$f \leftarrow (x+1)/(x-4)$


Γράψε f

Αλλιώς

Γράψε "αδύνατη"

Τέλος Αν

Τέλος Αν



Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M5. Πίνακες περιπτώσεων

π.χ. Μία εταιρεία ταχυδρομικών υπηρεσιών εφαρμόζει για τα έξοδα αποστολής ταχυδρομικών επιστολών εσωτερικού και εξωτερικού, χρέωση σύμφωνα με τον παρακάτω πίνακα:

Να γράψετε πρόγραμμα το οποίο:

α. Να διαβάζει το βάρος της επιστολής.

β. Να διαβάζει τον προορισμό της επιστολής. Η τιμή "ΕΣ" δηλώνει προορισμό εσωτερικού και η τιμή "ΕΞ" δηλώνει προορισμό εξωτερικού.

γ. Να υπολογίζει τα έξοδα αποστολής ανάλογα με τον προορισμό και το βάρος της επιστολής.

δ. Να εκτυπώνει τα έξοδα αποστολής.

Βάρος επιστολής σε γραμμάρια	Χρέωση εσωτερικού σε €	Χρέωση εξωτερικού σε €
από 0 έως και 500	2,0	4,8
από 500 έως και 1000	3,5	7,2
από 1000 και άνω	4,6	11,5

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M5. Πίνακες περιπτώσεων

Διάβασε β, πρ

Αν ($\beta \leq 500$) τότε

Αν (πρ = “ΕΣ”) τότε

$\kappa \leftarrow 2$

Αλλιώς

$\kappa \leftarrow 4.8$

ΤέλοςΑν

ΑλλιώςΑν ($\beta \leq 1000$) τότε

Αν (πρ = “ΕΣ”) τότε

$\kappa \leftarrow 3.5$

Αλλιώς

$\kappa \leftarrow 7.2$

ΤέλοςΑν

Αλλιώς

Αν (πρ = “ΕΣ”) τότε

$\kappa \leftarrow 4.6$

Αλλιώς

$\kappa \leftarrow 11.5$

ΤέλοςΑν

ΤέλοςΑν

Γράψε κ

Βάρος επιστολής σε γραμμάρια	Χρέωση εσωτερικού σε €	Χρέωση εξωτερικού σε €
από 0 έως και 500	2,0	4,8
από 500 έως και 1000	3,5	7,2
από 1000 και άνω	4,6	11,5


Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M6. Κλιμακωτή χρέωση

Π.χ. πρόγραμμα που διαβάζει το ετήσιο εισόδημα και τον αριθμό των παιδιών ενός φορολογούμενου και υπολογίζει κλιμακωτά τον φόρο του με βάση τα παρακάτω. Εάν η έκπτωση φόρου προκύψει μεγαλύτερη από τον φόρο, ο φόρος να μηδενίζεται.

Εισόδημα	Φόρος (%)
0-3000	0
3001-7000	10
7001-15000	14
15001-20000	20
20001-άνω	30

Παιδιά	Έκπτωση φόρου (€)
1	100
2	250
3	400
4 και άνω	550



Διάβασε ϵ , π Αν ($\epsilon \leq 3000$) τότε $\phi \leftarrow 0$ Αλλιώς Αν ($\epsilon \leq 7000$) τότε $\phi \leftarrow 10/100 * (\epsilon - 3000)$ Αλλιώς Αν ($\epsilon \leq 15000$) τότε $\phi \leftarrow 10/100 * 4000 + 14/100 * (\epsilon - 7000)$ Αλλιώς Αν ($\epsilon \leq 20000$) τότε $\phi \leftarrow 10/100 * 4000 + 14/100 * 8000 + 20/100 * (\epsilon - 15000)$ Αλλιώς $\phi \leftarrow 10/100 * 4000 + 14/100 * 8000 + 20/100 * 5000 + 30/100 * (\epsilon - 20000)$ Τέλος Αν	Αν ($\pi = 1$) τότε $\phi \leftarrow \phi - 100$ Αλλιώς Αν ($\pi = 2$) τότε $\phi \leftarrow \phi - 250$ Αλλιώς Αν ($\pi = 3$) τότε $\phi \leftarrow \phi - 400$ Αλλιώς Αν ($\pi \geq 4$) τότε $\phi \leftarrow \phi - 550$ Τέλος Αν Αν ($\phi < 0$) τότε $\phi \leftarrow 0$ Τέλος Αν Γράψε ϕ
--------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------	---------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M7. Διερεύνηση πραγματικού αριθμού

Π.χ. πρόγραμμα που διαβάζει έναν θετικό πραγματικό αριθμό και ελέγχει εάν είναι:

- ακέραιος, οπότε και τον χαρακτηρίζει ως άρτιο / περιττό
- πραγματικός, οπότε και τον στρογγυλοποιεί στον πλησιέστερο ακέραιο (π.χ. 7.3 → 7, 7.8 → 8, 7.5 → 8)

Διάβασε x

Αν $(x = A_M(x))$ τότε

Αν $(x \bmod 2 = 0)$ τότε

Γράψε "άρτιος"

Αλλιώς

Γράψε "περιττός"

ΤέλοςΑν

Αλλιώς

$\Delta M \leftarrow x - A_M(x)$

Αν $(\Delta M \geq 0.5)$ τότε

$\Sigma X \leftarrow A_M(x) + 1$

Αλλιώς

$\Sigma X \leftarrow A_M(x)$

ΤέλοςΑν

Γράψε ΣΧ

ΤέλοςΑν

ή
 $\Sigma X \leftarrow A_M(x+0.5)$

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M8. Μενού επιλογών

Π.χ. πρόγραμμα που με μενού επιλογών εκτελεί τις παρακάτω λειτουργίες:

1. Εμβαδό κύκλου 2. Εμβαδό τραπεζίου 3. Εμβαδό κυλίνδρου ($2\pi R^2 + 2\pi Rh$)

```
Γράψε "1. Εμβαδό κύκλου"  
Γράψε "2. Εμβαδό τραπεζίου"  
Γράψε "3. Εμβαδό κυλίνδρου"  
Γράψε "Διάλεξε (1-3)"  
Διάβασε επ  
Αν (επ = 1) τότε  
  Διάβασε ρ  
   $E \leftarrow 3.14 * \rho^2$   
  Γράψε E  
Αλλιώς Αν (επ = 2) τότε  
  Διάβασε β, Β, υ  
   $E \leftarrow (\beta + B) * \upsilon / 2$   
  Γράψε E  
Αλλιώς Αν (επ = 3) τότε  
  Διάβασε R, h  
   $E \leftarrow 3.14 * \rho^2 + 2 * 3.14 * R^2 + 2 * 3.14 * R * h$   
  Γράψε E  
Αλλιώς  
  Γράψε "Λάθος επιλογή"  
Τέλος Αν
```

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M9. Εφαρμογή των div και mod

Έστω ότι η χρέωση ενός κινητού είναι 0.30€ ανά πεντάλεπτο ομιλίας (η χρέωση ενός πενταλέπτου γίνεται ακόμη και αν δεν έχει εξαντληθεί). π.χ. για χρόνο ομιλίας 28 λεπτά, χρεώνονται 6 πεντάλεπτα. Εισαγωγή του χρόνου ομιλίας και εμφάνιση της χρέωσης (€)

Διάβασε t

$\lambda 5 \leftarrow t \text{ div } 5$

Αν $(t \text{ mod } 5 \neq 0)$ τότε

$\lambda 5 \leftarrow \lambda 5 + 1$

ΤέλοςΑν

$x \leftarrow 0.3 * \lambda 5$

Γράψε x

Εργάτης πληρώνεται προς 10€ την ώρα. Αν την τελευταία ώρα δουλέψει τουλάχιστον 30 λεπτά την πληρώνεται ολόκληρη διαφορετικά, δεν την πληρώνεται.

Εισαγωγή των συνολικών λεπτών εργασίας και υπολογισμός της αμοιβής του.

Διάβασε λ

$\omega \leftarrow \lambda \text{ div } 60$

Αν $(\lambda \text{ mod } 60 \geq 30)$ τότε

$\omega \leftarrow \omega + 1$

ΤέλοςΑν

Γράψε $\omega * 10$

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M9. Γενικά: «παρτίδες» των N για σύνολο X στοιχείων

$p \leftarrow X \text{ div } N$

Αν $(X \bmod N \neq 0)$ τότε

$p \leftarrow p + 1$

ΤέλοςΑν

ή

$p \leftarrow (X + N - 1) \text{ div } N$

ή

$p \leftarrow (X - 1) \text{ div } N + 1$

π.χ.

X : άτομα συμμετοχής σε εκδρομή

$N=50$ η χωρητικότητα ενός πούλμαν

p : απαραίτητος αριθμός πούλμαν

$p \leftarrow X \text{ div } 50$

Αν $(X \bmod 50 \neq 0)$ τότε

$p \leftarrow p + 1$

ΤέλοςΑν

ή

$p \leftarrow (X + 49) \text{ div } 50$

ή

$p \leftarrow (X - 1) \text{ div } 50 + 1$

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M10. Έλεγχος για δίσεκτο έτος

π.χ. πρόγραμμα που διαβάζει τον αριθμό ενός μήνα και ενός έτους και εμφανίζει τις ημέρες του μήνα (28/29/30/31). Όταν το έτος είναι δίσεκτο, ο Φεβρουάριος έχει 29 ημέρες, διαφορετικά έχει 28. Δίσεκτα είναι τα έτη που διαιρούνται με το 4 αλλά όχι με το 100, καθώς και εκείνα που διαιρούνται με το 400

```
Διάβασε μ, ε
Αν (μ = 2) τότε
  Αν (ε mod 4 = 0 ΚΑΙ ε mod 100 <> 0 Η ε mod 400 = 0) τότε
    Ημ ← 29
  Αλλιώς
    Ημ ← 28
ΤέλοςΑν
ΑλλιώςΑν (μ = 1 Η μ = 3 Η μ = 5 Η μ = 7 Η μ = 8 Η μ = 10 Η μ = 12) τότε
  Ημ ← 31
Αλλιώς
  Ημ ← 30
ΤέλοςΑν
Γράψε Ημ
```


Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M11. Πίνακας τιμών μεταβλητών

$a \leftarrow 7$

$\beta \leftarrow a^2$

$\gamma \leftarrow 5$

Αν $(\beta \bmod 2 = 1)$ **και** $(\gamma \geq 5)$ **τότε**

$\gamma \leftarrow \gamma + 2$

Αν $(\gamma > a)$ **τότε**

$a \leftarrow a^3$

$\gamma \leftarrow \gamma^2$

Αλλιώς

$a \leftarrow a * 4$

$\gamma \leftarrow \gamma * 2$

Τέλος_Αν

Γράψε a, β, γ

Τέλος_Αν

$a \leftarrow a \bmod \beta$

$\beta \leftarrow \beta \operatorname{div} \gamma$


$\gamma \leftarrow a \bmod \gamma$

Γράψε a, β, γ

α	β	γ	σθόνη
7	49	5	
7	49	7	
28	49	14	28, 49, 14
28	3	0	28, 3, 0

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M12. Μετατροπή: Αλγόριθμος \leftrightarrow ΔΡ



Πρόγραμμα ΔΡ
Μεταβλητές
Ακέραιες: X, Y
Αρχή
Διάβασε X
Αν (X mod 2 = 0) τότε
Y ← X div 2
Αν (Y <= 10) τότε
Y ← 2*X+Y
ΤέλοςΑν
Αλλιώς
Y ← X ^2
ΤέλοςΑν
Γράψε Y
ΤέλοςΠρογράμματος

M13. Ομόσημοι αριθμοί: $(x \geq 0 \text{ ΚΑΙ } y \geq 0) \text{ Η } (x < 0 \text{ ΚΑΙ } y < 0)$

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M14. Μετατροπή: Αν πολλαπλής επιλογής \leftrightarrow σύνολο από Αν-ΤέλοςΑν

```
Διάβασε x
Αν (x < 0) τότε
  x ← x + 10
ΑλλιώςΑν (x < 100) τότε
  x ← x + 5
Αλλιώς
  x ← x + 1
ΤέλοςΑν
```

```
Διάβασε x
ax ← x
Αν (ax < 0) τότε
  x ← x + 10
ΤέλοςΑν
Αν (ax >= 0 ΚΑΙ ax < 100) τότε
  x ← x + 5
ΤέλοςΑν
Αν (ax >= 100) τότε
  x ← x + 1
ΤέλοςΑν
```

```
Αν (x < 0) τότε
  Γράψε 1
Αλλιώς Αν (x < 1000) τότε
  Γράψε 2
Αλλιώς Αν (x < 2000 ΚΑΙ x <> 1500) τότε
  Γράψε 3
Αλλιώς
  Γράψε 4
ΤέλοςΑν
```

```
Αν (x < 0) τότε
  Γράψε 1
ΤέλοςΑν
Αν (x >= 0 ΚΑΙ x < 1000) τότε
  Γράψε 2
ΤέλοςΑν
Αν (x >= 1000 ΚΑΙ x < 2000 ΚΑΙ x <> 1500) τότε
  Γράψε 3
ΤέλοςΑν
Αν (x >= 2000 Η x = 1500) τότε
  Γράψε 4
ΤέλοςΑν
```

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M15. Κατάργηση του ΚΑΙ / Η

Αν (Σ1 ΚΑΙ Σ2) τότε
εντολές
ΤέλοςΑν

Αν (Σ1) τότε
Αν (Σ2) τότε
εντολές
ΤέλοςΑν
ΤέλοςΑν

Αν (Σ1 Η Σ2) τότε
εντολές
ΤέλοςΑν

Αν (Σ1) τότε
εντολές
ΑλλιώςΑν (Σ2) τότε
εντολές
ΤέλοςΑν

ή
Αν (Σ1) τότε
εντολές
ΤέλοςΑν
Αν (Σ2 ΚΑΙ ΟΧΙ Σ1) τότε
εντολές
ΤέλοςΑν

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

M15. Κατάργηση του ΚΑΙ / Η

Αν (Σ1 ΚΑΙ Σ2) τότε
εντολές1
Αλλιώς
εντολές2
ΤέλοςΑν

Αν (Σ1) τότε
Αν (Σ2) τότε
εντολές1
Αλλιώς
εντολές2
ΤέλοςΑν
Αλλιώς
εντολές2
ΤέλοςΑν

Αν (Σ1 Η Σ2) τότε
εντολές1
Αλλιώς
εντολές2
ΤέλοςΑν

Αν (Σ1) τότε
εντολές1
ΑλλιώςΑν (Σ2) τότε
εντολές1
Αλλιώς
εντολές2
ΤέλοςΑν

ή
Αν (Σ1) τότε
εντολές1
ΤέλοςΑν
Αν (Σ2 ΚΑΙ ΟΧΙ Σ1) τότε
εντολές1
ΤέλοςΑν
Αν (ΟΧΙ Σ1 ΚΑΙ ΟΧΙ Σ2) τότε
εντολές2
ΤέλοςΑν

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

Διάφορα (απλούστευση λογικών εκφράσεων)

Δώστε ισοδύναμες λογικές τιμές (Αληθής/Ψευδής) ή απλούστερες λογικές εκφράσεις για τα παρακάτω:

$$x \geq 9 \text{ ΚΑΙ } x < 7$$

Ψευδής

$$y <> 5 \text{ Η } y <> 8$$

Αληθής

$$z <> 11 \text{ Η } z \geq -1$$

Αληθής

$$\varphi = 1 \text{ Η } \varphi <> 1$$

Αληθής

$$\lambda = 1 \text{ Η } 2 = 2$$

Αληθής

$$\mu > 0 \text{ ΚΑΙ } 2 > 3$$

Ψευδής

$$\rho > 1 \text{ ΚΑΙ } \rho \geq 4$$

$\rho \geq 4$

$$\tau < 0 \text{ ΚΑΙ } \tau < 7$$

$\tau < 0$

$$\text{ΟΧΙ } \Sigma \iff \Sigma = \text{Ψευδής} \iff \Sigma <> \text{Αληθής}$$

$$x \geq y \text{ ΚΑΙ } y \leq x \iff x = y$$

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

Διάφορα (έκφραση → λογική συνθήκη, συμπλήρωση κενών)

Έστω u το ύψος ενός ατόμου. Ποιες είναι οι αντίστοιχες λογικές συνθήκες;

όσοι έχουν ύψος μεταξύ του 1.80 και του 1.90

$u > 1.80$ ΚΑΙ $u < 1.90$

όσοι έχουν ύψος από 1.70 έως και 1.85

$u \geq 1.70$ ΚΑΙ $u \leq 1.85$

όσοι έχουν ύψος κάτω του 1.80 και
όσοι έχουν ύψος άνω του 1.90

$u < 1.80$ Η $u > 1.90$

Συμπληρώστε τα κενά με Αληθής/Ψευδής. Εάν ταιριάζουν και τα δύο, συμπληρώστε με τη λέξη Οτιδήποτε

1. Ψευδής Η ... = Αληθής

5. Αληθής ΚΑΙ ... = Ψευδής

2. Αληθής Η ... = ...

6. Αληθής Η Ψευδής ΚΑΙ Αληθής = ...

3. Ψευδής ΚΑΙ ... = ...

7. ΟΧΙ Ψευδής ΚΑΙ Αληθής Η ... = ...

4. Αληθής ΚΑΙ ... = Αληθής

1. Α 2. Ο, Α 3. Ο, Ψ 4. Α 5. Ψ 6. Α 7. Ο, Α

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

Διάφορα (Αν πολλαπλής επιλογής → διαστήματα τιμών)

Διάβασε x

Αν $(x < -5 \text{ Η } x > 80 \text{ ΚΑΙ } x \leq 140)$ τότε

Γράψε 1

Αλλιώς Αν $(x > 40 \text{ ΚΑΙ } x \leq 200)$ τότε

Γράψε 2

Αλλιώς

Γράψε 3


Αν $(x > 200)$ τότε

Γράψε 4

Τέλος Αν

Τέλος Αν

Για ποιες τιμές του x γράφει τις τιμές 1, 2, 3 και 4;



$$1: x \in (-\infty, -5) \cup (80, 140]$$


$$2: x \in (40, 80] \cup (140, 200]$$

$$3: x \in [-5, 40] \cup (200, \infty)$$

$$4: x \in (200, \infty)$$

Δομή επιλογής – Μεθοδολογίες Ασκήσεων

Διάφορα (απαλοιφή κώδικα "κοινού παράγοντα" σε δομή Αν)



Δομή επιλογής – Μεθοδολογίες Ασκήσεων

Διάφορα (ισοδύναμος κώδικας)

Αν $(x \bmod 2 = 0)$ τότε
Γράψε x
Αλλιώς
Γράψε $-x$
ΤέλοςΑν



Γράψε $(-1)^{x*x}$

ή

Γράψε $(-1)^{(2 - x \bmod 2)*x}$

Αν $(x \bmod 2 = 0)$ τότε
Γράψε 100
Αλλιώς
Γράψε 99
ΤέλοςΑν



Γράψε $100 - A_T(x \bmod 2)$

Αν $(x \bmod 2 = 0)$ τότε
Γράψε 1
Αλλιώς
Γράψε x
ΤέλοςΑν



Γράψε $x^{(x \bmod 2)}$

Οι ακέραιοι x, y είναι
και οι δύο άρτιοι ή και
οι δύο περιττοί



$x \bmod 2 = 0$ ΚΑΙ $y \bmod 2 = 0$ Η $x \bmod 2 \neq 0$ ΚΑΙ $y \bmod 2 \neq 0$



$x \bmod 2 = y \bmod 2$




$(x + y) \bmod 2 = 0$

(§ 2.4.5 - § 8.2) Δομή Επανάληψης

1. Δομή Όσο - επανάλαβε

Σύνταξη

```
ΟΣΟ συνθήκη ΕΠΑΝΑΛΑΒΕ  
  εντολή-1  
  εντολή-2  
  ...  
  εντολή-ν  
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```



Λειτουργία

Ελέγχεται η συνθήκη και αν είναι Αληθής, εκτελούνται οι εντολές που βρίσκονται ανάμεσα στις **ΟΣΟ_ΕΠΑΝΑΛΑΒΕ** και **ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ**. Στη συνέχεια ελέγχεται πάλι η συνθήκη και αν ισχύει, εκτελούνται πάλι οι ίδιες εντολές. Όταν η λογική έκφραση γίνει Ψευδής, τότε σταματάει η επανάληψη και εκτελείται η εντολή μετά το **ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ**.

αριθμός επαναλήψεων $[0, \infty)$, αριθμός ελέγχου της συνθήκης $[1, \infty)$

αριθμός ελέγχου της συνθήκης = αριθμός επαναλήψεων + 1

περιπτώσεις χρήσης (άγνωστο πλήθος επαναλήψεων που μπορεί να είναι και 0).

π.χ. 1	π.χ. 2	π.χ. 3
$\alpha \leftarrow 5$ Όσο ($\alpha \leq 11$) επανάλαβε $\alpha \leftarrow \alpha + 2$ Τέλος Επανάληψης	$\alpha \leftarrow -3$ Όσο ($\alpha > -1$) επανάλαβε $\alpha \leftarrow \alpha + 4$ Τέλος Επανάληψης	$\alpha \leftarrow 5$ Όσο ($\alpha < 10$) επανάλαβε $\alpha \leftarrow \alpha + 2$ Τέλος Επανάληψης

π.χ. 1	π.χ. 2	π.χ. 3
4 επαναλήψεις	0 επαναλήψεις	∞ επαναλήψεις


(§ 2.4.5 - § 8.2) Δομή Επανάληψης

1. Δομή Όσο - επανάλαβε

Παράδειγμα:

Πρόγραμμα που διαβάζει ένα αρχικό κεφάλαιο κατάθεσης σε τράπεζα με ετήσιο επιτόκιο 4%. Εμφανίζει σε πόσα χρόνια το κεφάλαιο θα διπλασιασθεί.

```
Διάβασε κ  
έτη ← 0  
στόχος ← 2 * κ  
Όσο (κ < στόχος) επανάλαβε  
  έτη ← έτη + 1  
  κ ← κ + 4/100 * κ  
Τέλος Επανάληψης  
Γράψε έτη
```



(§ 2.4.5 - § 8.2) Δομή Επανάληψης

2. Δομή ΜέχριςΌτου

Σύνταξη

ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ


εντολή-1

εντολή-2

...

εντολή-n

ΜΕΧΡΙΣ_ΟΤΟΥ λογική-έκφραση



Λειτουργία

Εκτελούνται οι εντολές μεταξύ των **ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ** και **ΜΕΧΡΙΣ_ΟΤΟΥ**. Στη συνέχεια ελέγχεται η λογική έκφραση και αν δεν ισχύει (είναι ψευδής), τότε οι εντολές που βρίσκονται ανάμεσα στις **ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ** και **ΜΕΧΡΙΣ_ΟΤΟΥ**, εκτελούνται πάλι. Ελέγχεται ξανά η λογική έκφραση και αν δεν ισχύει, επαναλαμβάνεται η εκτέλεση των ίδιων εντολών.

Όταν η λογική έκφραση γίνει Αληθής τότε σταματάει η επανάληψη και εκτελείται η εντολή μετά από την **ΜΕΧΡΙΣ_ΟΤΟΥ**.

αριθμός επαναλήψεων $[1, \infty)$, αριθμός ελέγχου της συνθήκης $[1, \infty)$
αριθμός ελέγχου της συνθήκης = αριθμός επαναλήψεων
περιπτώσεις χρήσης (άγνωστο πλήθος επαναλήψεων που είναι τουλάχιστον 1).

π.χ. 1	π.χ. 2	π.χ. 3
$\alpha \leftarrow 5$ ΑρχήΕπανάληψης $\alpha \leftarrow \alpha + 2$ ΜέχριςΌτου ($\alpha > 11$)	$\alpha \leftarrow -3$ ΑρχήΕπανάληψης $\alpha \leftarrow \alpha + 4$ ΜέχριςΌτου ($\alpha \geq 1$)	$\alpha \leftarrow 5$ ΑρχήΕπανάληψης $\alpha \leftarrow \alpha + 2$ ΜέχριςΌτου ($\alpha = 12$)

π.χ. 1	π.χ. 2	π.χ. 3
4 επαναλήψεις	1 επανάληψη	∞ επαναλήψεις


(§ 2.4.5 - § 8.2) Δομή Επανάληψης

2. Δομή ΜέχριςΌτου

Παράδειγμα:

Πρόγραμμα που διαβάζει επαναληπτικά θετικές ακέραιες τιμές και τις χαρακτηρίζει ως άρτιο/περιττό. Να σταματάει όταν δοθεί η τιμή μηδέν(0).

```
ΑρχήΕπανάληψης  
Διάβασε α  
Αν (α mod 2 = 0) τότε  
  Γράψε " άρτιος"  
Αλλιώς  
  Γράψε " περιττός"  
ΤέλοςΑν  
ΜέχριςΌτου (α = 0)
```



(§ 2.4.5 - § 8.2) Δομή Επανάληψης

3. Δομή Για-από-μέχρι



Σύνταξη:

για <μετρητής> από <αρχική> μέχρι <τελική> [μεβήμα β]

<εντολές>

τέλος Επανάληψης

- όταν το βήμα δεν αναγράφεται, εννοείται το 1





Εντός της Για **δεν** επιτρέπεται η τροποποίηση της τιμής του μετρητή

(§ 2.4.5 - § 8.2) Δομή Επανάληψης

3. Δομή Για-από-μέχρι

Λειτουργία:

Ο ΗΥ “σαρώνει” μέσω του μετρητή το διάστημα:



και σε κάθε επανάληψη, εκτελεί τις εντολές.

Πιθανός αριθμός επαναλήψεων: $[0, +\infty)$

$\beta = 0 \Rightarrow \infty$ επαναλήψεις

$\beta \neq 0$ και αρχική = τελική $\Rightarrow 1$ επανάληψη

$\beta > 0$ και αρχική > τελική ή

$\beta < 0$ και αρχική < τελική $\Rightarrow 0$ επαναλήψεις

σε κάθε άλλη περίπτωση, επαναλήψεις =

$$A_M\left(\left\lceil \frac{|\text{τελική} - \text{αρχική}|}{\beta\text{ήμα}} \right\rceil + 1\right)$$

Χρήση: όταν έχω επανάληψη **ΓΝΩΣΤΟΥ** πλήθους επαναλήψεων.

(§ 2.4.5 - § 8.2) Δομή Επανάληψης

3. Δομή Για-από-μέχρι, παραδείγματα

Εμφάνιση των 100 πρώτων ακεραίων:
για x από 1 μέχρι 100

Γράψε x

ΤέλοςΕπανάληψης

Εμφάνιση των αριθμών: 50, 49, ... , 32, 31, 30
για y από 50 μέχρι 30 μεβήμα -1

Γράψε y

ΤέλοςΕπανάληψης

Εμφάνιση των αριθμών: 1.7, 1.8, ... , 9.7, 9.8
για z από 1.7 μέχρι 9.8 μεβήμα 0.1

Γράψε z

ΤέλοςΕπανάληψης

Εμφάνιση της λέξης "Καλημέρα" χίλιες φορές
για κ από 1 μέχρι 1000

Γράψε "Καλημέρα"

ΤέλοςΕπανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης

3. Δομή Για-από-μέχρι, παραδείγματα

Εμφάνιση των 100 πρώτων πολλαπλασίων του 7 :
για κ από 7 μέχρι 700 μεβήμα 7

Γράψε κ
ΤέλοςΕπανάληψης

Υπολογισμός του αθροίσματος $S = 5 + 10 + 15 + \dots + 500$:

$S \leftarrow 0$

για x από 5 μέχρι 500 μεβήμα 5

$S \leftarrow S + x$

ΤέλοςΕπανάληψης

Γράψε S

Υπολογισμός του γινομένου $\Gamma = 3 \times 6 \times 9 \times 12 \times \dots \times 300$:

$\Gamma \leftarrow 1$

για x από 3 μέχρι 300 μεβήμα 3

$\Gamma \leftarrow \Gamma * x$

ΤέλοςΕπανάληψης

Γράψε Γ

(§ 2.4.5 - § 8.2) Δομή Επανάληψης

3. Δομή Για-από-μέχρι, παραδείγματα

Υπολογισμός του αθροίσματος $S = 1^2 + 3^2 + 5^2 + \dots + 99^2$:

$S \leftarrow 0$

για x από 1 μέχρι 99 μεβήμα 2

$S \leftarrow S + x^2$

ΤέλοςΕπανάληψης

Γράψε S

Υπολογισμός του a^β , $\beta > 0$, $\beta \in \mathbb{Z}$:

Διάβασε a , β

$\Delta \leftarrow 1$

για x από 1 μέχρι β

$\Delta \leftarrow \Delta * a$

ΤέλοςΕπανάληψης

Γράψε Δ

(§ 2.4.5 - § 8.2) Δομή Επανάληψης

3. Δομή Για-από-μέχρι, παραδείγματα

Υπολογισμός της παράστασης:

$$S = \frac{1^2 + 3^2 + 5^2 + \dots + 99^2}{2^3 + 4^3 + 6^3 + \dots + 100^3}$$

S1 ← 0

S2 ← 0

για x από 1 μέχρι 99 μεβήμα 2

S1 ← S1 + x²

S2 ← S2 + (x+1)³

ΤέλοςΕπανάληψης

S ← S1 / S2

Γράψε S

Υπολογισμός της παράστασης: $S = 1^{10} + 2^9 + 3^8 + \dots + 9^2 + 10^1$

S ← 0

Δ ← 10

για x από 1 μέχρι 10

S ← S + x^Δ

Δ ← Δ - 1

ΤέλοςΕπανάληψης

Γράψε S

(§ 2.4.5 - § 8.2) Δομή Επανάληψης

3. Δομή Για-από-μέχρι, παραδείγματα

ποιό το πλήθος των επαναλήψεων και τί εμφανίζουν:

1. για κ από 5 μέχρι 35 μεβήμα 7, Γράψε κ
2. για λ από 14 μέχρι 31 μεβήμα 0 , Γράψε λ
3. για μ από 15 μέχρι 4 μεβήμα 2 , Γράψε μ
4. για ν από 11 μέχρι 23 μεβήμα -3 , Γράψε ν
5. για λ από 11 μέχρι 11 μεβήμα 4 , Γράψε λ
6. για x από 314 μέχρι 5729 μεβήμα 7 (τύπος \Rightarrow 774 επαναλήψεις)
7. $T \leftarrow 1$; για x από 1 μέχρι T; Γράψε x; $T \leftarrow T + 1$; ΤέλοςΕπανάληψης $\Rightarrow \infty$

ποιό το πλήθος των επαναλήψεων και τί εμφανίζει:

για x από 1 μέχρι 9 μεβήμα 2

για y από 11 μέχρι 2 μεβήμα -3

Γράψε x, y

ΤέλοςΕπανάληψης

ΤέλοςΕπανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης

3. Δομή Για-από-μέχρι, παραδείγματα

Εμφάνιση όλων των συνδυασμών ρίψης 2 ζαριών:

για x από 1 μέχρι 6

για y από 1 μέχρι 6

Γράψε x, y

ΤέλοςΕπανάληψης

ΤέλοςΕπανάληψης

Εμφάνιση όλων των ενδείξεων ενός ψηφιακού ρολογιού από την ώρα 0:0:0 έως και 23:59:59:

για ω από 0 μέχρι 23

για λ από 0 μέχρι 59

για δ από 0 μέχρι 59

Γράψε ω, ":", λ, ":", δ

ΤέλοςΕπανάληψης

ΤέλοςΕπανάληψης

ΤέλοςΕπανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης

3. Δομή Για-από-μέχρι, παραδείγματα


Πότε τερματίζεται το παρακάτω:
για x από 1 μέχρι 100

 Διάβασε x
ΤέλοςΕπανάληψης

Όταν και εφόσον δοθεί τιμή > 99
(γενικά όμως: εντός της για δεν
επιτρέπεται η τροποποίηση του μετρητή)

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M1. Μετατροπές δεδομένου επαναληπτικού κώδικα: Όσο \leftrightarrow Μέχρι \leftrightarrow Για



Για \rightarrow Όσο		
για x από 1 μέχρι 100 Γράψε x ΤέλοςΕπ.	\Rightarrow	x \leftarrow 1 Όσο (x \leq 100) επ. Γράψε x x \leftarrow x + 1 ΤέλοςΕπ.

Για \rightarrow Μέχρι
Για \rightarrow Όσο \rightarrow Μέχρι

Όσο \rightarrow Μέχρι		
Όσο (Σ) επ. εντολές ΤέλοςΕπ	\Rightarrow	Αν (Σ) τότε ΑρχήΕπανάληψης εντολές ΜέχριςΌτου (ΟΧΙ Σ) ΤέλοςΑν

Μέχρι \rightarrow Όσο		
ΑρχήΕπανάληψης εντολές ΜέχριςΌτου (Σ)	\Rightarrow	εντολές Όσο (ΟΧΙ Σ) επ. εντολές ΤέλοςΕπ

Μέχρι \rightarrow Για
Μέχρι \rightarrow Όσο \rightarrow Για

Όσο / Μέχρι \rightarrow Για
<p>Η μετατροπή είναι δυνατή, εάν ισχύουν τα παρακάτω:</p> <ul style="list-style-type: none"> • μία μεταβλητή ελέγχου (μετρητής) • γνωστή αρχική τιμή (από) • γνωστή τελική τιμή (μέχρι) • γνωστή και σταθερή μετατροπή του μετρητή (βήμα)

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M1. Μετατροπές δεδομένου επαναληπτικού κώδικα: Όσο \rightarrow Για

$\mu \leftarrow$ αρχική

Όσο ($\mu \leq$ τελική) επανάλαβε ! ή \geq

εντολές1

$\mu \leftarrow \mu + \beta$! ή $\mu \leftarrow \mu - \beta$

εντολές2

ΤέλοςΕπανάληψης

Μετατροπή σε για



για μ από αρχική μέχρι τελική μεβήμα β

εντολές1 ! όπου $\mu: \mu$

εντολές2 ! όπου $\mu: \mu + \beta$ ή $\mu - \beta$

ΤέλοςΕπανάληψης

π.χ.

$\mu \leftarrow 5$

Όσο ($\mu \leq 100$) επανάλαβε

Γράψε μ

$\mu \leftarrow \mu + 7$

Γράψε μ

ΤέλοςΕπανάληψης

Μετατροπή σε για



για μ από 5 μέχρι 100 μεβήμα 7

Γράψε μ

Γράψε $\mu + 7$

ΤέλοςΕπανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M1. Μετατροπές δεδομένου επαναληπτικού κώδικα: Όσο \rightarrow Για

$x \leftarrow 1$

Όσο ($x < 10$) επανάλαβε

εντολές

$x \leftarrow x + 1$

ΤέλοςΕπανάληψης

Μετατροπή σε για



για x από 1 μέχρι 9

εντολές

ΤέλοςΕπανάληψης

$x \leftarrow 1.8$

Όσο ($x < 10$) επανάλαβε

εντολές

$x \leftarrow x + 1$

ΤέλοςΕπανάληψης

Μετατροπή σε για



για x από 1.8 μέχρι 9.8

εντολές

ΤέλοςΕπανάληψης

$x \leftarrow 9.3$

Όσο ($x > 0$) επανάλαβε

εντολές

$x \leftarrow x - 1$

ΤέλοςΕπανάληψης

Μετατροπή σε για



για x από 9.3 μέχρι 0.3 μεβήμα -1

εντολές

ΤέλοςΕπανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M2. Στατιστικά στοιχεία: ΜΟ, Sum, min, max, πλήθη, % ποσοστά

α) γνωστού πλήθους στοιχείων

π.χ. Πρόγραμμα που για τους 500 υπαλλήλους μιας εταιρείας, διαβάζει το μισθό τους και τον τομέα τους (Α/Β/Γ). Εμφανίζει: τον μέσο μισθό της εταιρείας, τους μέσους μισθούς ανά τομέα, τον μικρότερο και τον μεγαλύτερο μισθό, το % ποσοστό των υψηλόμισθων (> 2000 €)

```
S ← 0 SA ← 0 SB ← 0 SΓ ← 0
ΠΑ ← 0 ΠΒ ← 0 ΠΓ ← 0 ΠΥ ← 0
```

για i από 1 μέχρι 500

Διάβασε μ, τ

S ← S + μ

Αν (τ = "Α") τότε

SA ← SA + μ

ΠΑ ← ΠΑ + 1

Αλλιώς Αν (τ = "Β") τότε

SB ← SB + μ

ΠΒ ← ΠΒ + 1

Αλλιώς

SΓ ← SΓ + μ

ΠΓ ← ΠΓ + 1

ΤέλοςΑν

Αν (μ > 2000) τότε

ΠΥ ← ΠΥ + 1

ΤέλοςΑν

Αν (i = 1) τότε

min ← μ

max ← μ

Αλλιώς

Αν (μ < min) τότε

min ← μ

ΤέλοςΑν

Αν (μ > max) τότε

max ← μ

ΤέλοςΑν

ΤέλοςΑν

ΤέλοςΕπανάληψης

μο ← S / 500

Γράψε μο, min, max, ΠΥ/500*100, "%"

Αν (ΠΑ <> 0) τότε

Γράψε SA / ΠΑ

ΤέλοςΑν

! το ίδιο για ΠΒ, ΠΓ

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M2. Στατιστικά στοιχεία: ΜΟ, Sum, min, max, πλήθη, % ποσοστά

β) αγνώστου πλήθους στοιχείων

π.χ1. Πρόγραμμα που διαβάζει επαναληπτικά τους βαθμούς μαθητών σε 1 διαγώνισμα (0-20) μέχρι να δοθεί ως βαθμός η τιμή -1. Εμφανίζει: τον μέσο όρο βαθμολογίας των μαθητών, τον μικρότερο και τον μεγαλύτερο βαθμό, το % ποσοστό των άριστων (> 18)

$\Sigma \leftarrow 0$ $\Pi \leftarrow 0$ $\Pi\Lambda \leftarrow 0$

$\text{min} \leftarrow 21$ $\text{max} \leftarrow -1$

Διάβασε β

Όσο (β <> -1) επανάλαβε

$\Sigma \leftarrow \Sigma + \beta$

$\Pi \leftarrow \Pi + 1$

Αν (β > max) τότε

$\text{max} \leftarrow \beta$

ΤέλοςΑν

Αν (β < min) τότε

$\text{min} \leftarrow \beta$

ΤέλοςΑν

Αν (β > 18) τότε

$\Pi\Lambda \leftarrow \Pi\Lambda + 1$

ΤέλοςΑν

Διάβασε β

ΤέλοςΕπανάληψης

Αν (Π <> 0) τότε

$\text{ΜΟ} \leftarrow \Sigma / \Pi$

Γράψε ΜΟ, min, max, $\Pi\Lambda/\Pi*100$, '%'

Αλλιώς

Γράψε 'Δεν δόθηκαν βαθμοί'

ΤέλοςΑν

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M2. Στατιστικά στοιχεία: MO, Sum, min, max, πλήθη, % ποσοστά

β) αγνώστου πλήθους στοιχείων

π.χ2. Πρόγραμμα που διαβάζει επαναληπτικά τους βαθμούς μαθητών σε 1 διαγώνισμα (0-20) μέχρι να δοθεί ως βαθμός η τιμή -1. Εμφανίζει τους 2 μεγαλύτερους βαθμούς

```
max1 ← -1 max2 ← -1
```

Διάβασε β

Όσο (β <> -1) επανάλαβε

Αν (β > max1) τότε

```
max2 ← max1
```

```
max1 ← β
```

Αλλιώς Αν (β > max2) τότε

```
max2 ← β
```

Τέλος Αν

Διάβασε β

Τέλος Επανάληψης

Αν (max1 <> -1) τότε

Γράψε '1^{ος} μεγαλύτερος:', max1

Τέλος Αν

Αν (max2 <> -1) τότε

Γράψε '2^{ος} μεγαλύτερος:', max2

Τέλος Αν

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M2. Στατιστικά στοιχεία: MO, Sum, min, max, πλήθη, % ποσοστά

β) αγνώστου πλήθους στοιχείων

π.χ3. Πρόγραμμα που διαβάζει επαναληπτικά τους βαθμούς μαθητών σε 1 διαγώνισμα (0-20) μέχρι να δοθεί ως βαθμός η τιμή -1. Εμφανίζει το μεγαλύτερο βαθμό και το πλήθος των βαθμών που είναι ίσοι με το αυτόν.

max ← -1

Πmax ← 0

Διάβασε β

Όσο (β <> -1) επανάλαβε

Αν (β > max) τότε

max ← β

Πmax ← 1

Αλλιώς Αν (β = max) τότε

Πmax ← Πmax + 1

Τέλος Αν

Διάβασε β

Τέλος Επανάληψης

Αν (max <> -1) τότε

Γράψε '1^{ος} μεγαλύτερος:', max, Πmax

Τέλος Αν

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M2. Στατιστικά στοιχεία: ΜΟ, Sum, min, max, πλήθη, % ποσοστά

β) αγνώστου πλήθους στοιχείων

π.χ.4 Εισαγωγή ονομάτων και μισθών για τους υπαλλήλους μιας εταιρείας. Να διακόπτεται όταν δοθεί σαν όνομα η τιμή 'τέλος' ή δοθούν 10 υψηλόμισθοι (>2000€). Εμφάνιση του πιο καλά αμειβόμενου υπαλλήλου καθώς και τον μισθό του (μόνο ένας).

max ← -1 οnmax ← '' ΠΥ ← 0

Διάβασε ον

Όσο (ον <> 'τέλος' ΚΑΙ ΠΥ < 10) επανάλαβε

Διάβασε μ

Αν (μ > 2000) τότε

ΠΥ ← ΠΥ + 1

ΤέλοςΑν

Αν (μ > max) τότε

max ← μ

οnmax ← ον

ΤέλοςΑν

Διάβασε ον

ΤέλοςΕπανάληψης

Αν (max <> -1) τότε

Γράψε οnmax, max

ΤέλοςΑν

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M3. Τύποι μετατροπής για ένα διάστημα τιμών

π.χ. Πρόγραμμα που για κάθε θερμοκρασία κελσίου $[-100, 100]$ μεβήμα 0.1, εμφανίζει την αντίστοιχη θερμοκρασία Fahrenheit $((F-32)/9 = C/5)$

για C από -100 μέχρι 100 μεβήμα 0.1

$F \leftarrow 9 * C / 5 + 32$

Γράψε F

Τέλος Επανάληψης

M4. Έλεγχος εγκυρότητας δεδομένων

π.χ. Εισαγωγή βαθμού στην 20-θμια κλίμακα με έλεγχο εγκυρότητας

1 ^{ος} τρόπος	2 ^{ος} τρόπος
Αρχή Επανάληψης	Διάβασε β
Διάβασε β	Όσο $(\beta < 0 \vee \beta > 20)$ επανάλαβε
Μέχρις Ότου $(\beta \geq 0 \wedge \beta \leq 20)$	Γράψε “Λάθος τιμή”
	Διάβασε β
	Τέλος Επανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M5. Ποσότητα που μεταβάλλεται

α) για γνωστό πλήθος επαναλήψεων

π.χ. Πρόγραμμα που διαβάζει ένα αρχικό κεφάλαιο κατάθεσης σε τράπεζα και το % ετήσιο επιτόκιο της. Εμφανίζει το τελικό κεφάλαιο μετά από 30 χρόνια.

Διάβασε κ, επ

για i από 1 μέχρι 30

$k \leftarrow k + επ/100 * k$

ΤέλοςΕπανάληψης

Γράψε κ

β) για άγνωστο πλήθος επαναλήψεων

π.χ.1 Έστω ότι σήμερα ζουν 40.000 θαλάσσιες χελώνες και ετησίως μειώνονται κατά 5,7%. Πρόγραμμα που υπολογίζει σε πόσα χρόνια θα γίνουν λιγότερες από 10.000.

$X \leftarrow 40000$

έτη $\leftarrow 0$

Όσο ($X \geq 10000$) επανάλαβε

$X \leftarrow X - 5.7/100 * X$

έτη \leftarrow έτη + 1

ΤέλοςΕπανάληψης

Γράψε έτη

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M5. Ποσότητα που μεταβάλλεται

β) για άγνωστο πλήθος επαναλήψεων (2 ποσότητες)

π.χ.2 Δύο δέντρα έχουν σήμερα ύψος 2m και 3m αντίστοιχα. Ετησίως ψηλώνουν και 5% και 4% αντίστοιχα. Πρόγραμμα που υπολογίζει σε πόσα χρόνια το 1ο δέντρο θα ξεπεράσει σε ύψος το 2ο και τι ύψος θα έχουν τότε.

```
ΠΡΟΓΡΑΜΜΑ trees2
ΜΕΤΑΒΛΗΤΕΣ
  ΠΡΑΓΜΑΤΙΚΕΣ: h1, h2, pos1, pos2
  ΑΚΕΡΑΙΕΣ: eth
ΑΡΧΗ
  h1 <-- 2
  pos1 <-- 5
  h2 <-- 3
  pos2 <-- 4
  eth <-- 0
  ΟΣΟ h1 <= h2 ΕΠΑΝΑΛΑΒΕ
    h1 <-- h1 + pos1/100*h1
    h2 <-- h2 + pos2/100*h2
    eth <-- eth + 1
  ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
  ΓΡΑΨΕ 'Σε ',eth,'χρόνια το 1ο δέντρο:',h1,'το 2ο δέντρο:',h2
ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ
```

Σε 43 χρόνια το 1ο δέντρο: 16.30 το 2ο δέντρο: 16.20

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M6. Ερώτηση προς τον χρήστη για συνέχεια (ΝΑΙ/ΟΧΙ)

π.χ. Πρόγραμμα που διαβάζει επαναληπτικά βαθμούς απολυτηρίων και τους χαρακτηρίζει ως: άριστος (>18) / πολύ καλός (>16) / άλλος. Σε κάθε επανάληψη να ρωτάει τον χρήστη εάν επιθυμεί να συνεχίσει (ΝΑΙ/ΟΧΙ).

ΑρχήΕπανάληψης

Διάβασε β

Αν ($\beta > 18$) τότε

Γράψε "άριστος"

ΑλλιώςΑν ($\beta > 16$) τότε

Γράψε "πολύ καλός"

Αλλιώς

Γράψε "άλλος"

ΤέλοςΑν


Γράψε "Συνέχεια; (ΝΑΙ/ΟΧΙ)"

Διάβασε απ


ΜέχριςΌτου(απ = "ΟΧΙ")

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων


M7. Μετατροπή : Αλγόριθμος \leftrightarrow ΔΡ




Αν (Συνθήκη) τότε
εντολές
ΤέλοςΑν



Αν (Συνθήκη) τότε
εντολές 1
Αλλιώς
εντολές 2
ΤέλοςΑν




Αν (Συνθήκη 1) τότε
εντολές 1
ΑλλιώςΑν (Συνθήκη 2) τότε
εντολές 2
...
ΑλλιώςΑν (Συνθήκη ν) τότε
εντολές ν
Αλλιώς
εντολές ν+1
ΤέλοςΑν




Αν (Συνθήκη 1) τότε
εντολές 1
ΑλλιώςΑν (Συνθήκη 2) τότε
εντολές 2
...
ΑλλιώςΑν (Συνθήκη ν) τότε
εντολές ν
ΤέλοςΑν

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων


M7. Μετατροπή : Αλγόριθμος \leftrightarrow ΔΡ




Όσο (Συνθήκη) επανάλαιβε
εντολές
Τέλος Επανάληψης



Αρχή Επανάληψης
Εντολές
Μέχρις Ότου (Συνθήκη)




Για μ από α μέχρι τ μεβήμα β ! β > 0
Εντολές
Τέλος Επανάληψης




Για μ από α μέχρι τ μεβήμα β ! β < 0
Εντολές
Τέλος Επανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων


M7. Μετατροπή : Αλγόριθμος ↔ ΔΡ



Αν (OXI Συνθήκη) τότε
εντολές
ΤέλοςΑν




Όσο (OXI Συνθήκη) επανάλαβε
εντολές
ΤέλοςΕπανάληψης



ΑρχήΕπανάληψης
εντολές
ΜέχριςΌτου (OXI Συνθήκη)

ή


εντολές
Όσο (Συνθήκη) επανάλαβε
εντολές
ΤέλοςΕπανάληψης




εντολές 1
Όσο (Συνθήκη) επανάλαβε
εντολές 2
εντολές 1
ΤέλοςΕπανάληψης

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M7. Μετατροπή : Αλγόριθμος \leftrightarrow ΔΡ




stop \leftarrow Ψευδής
Όσο (Σ1 ΚΑΙ stop = Ψευδής) επανάλαβε
 εντολές1
 Αν (Σ2) τότε
 εντολές2
 Αλλιώς
 εντολές3
 stop \leftarrow Αληθής
ΤέλοςΑν
ΤέλοςΕπανάληψης



Πρόγραμμα ΔΡ
Μεταβλητές
 Ακέραιες: A, B, S
Αρχή
 Διάβασε A, B
 S \leftarrow 0
 Όσο (B > 0) επανάλαβε
 Αν (B mod 2 = 1) τότε
 S \leftarrow S + A
 ΤέλοςΑν
 A \leftarrow A * 2
 B \leftarrow B div 2
 ΤέλοςΕπανάληψης
 Γράψε S
ΤέλοςΠρογράμματος

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M7. Μετατροπή : Αλγόριθμος \leftrightarrow ΔΡ



Αλγόριθμος Μετατροπή

Αρχή επανάληψης

Διάβασε πλήθος

Μέχρις ότου πλήθος > 0

άθροισμα $\leftarrow 0$

$i \leftarrow 1$

Όσο $i \leq$ πλήθος επανάλαβε

Διάβασε αριθμός

άθροισμα \leftarrow άθροισμα + αριθμός

$i \leftarrow i + 1$

Τέλος επανάληψης

Εμφάνισε άθροισμα

Τέλος Μετατροπή

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M7. Μετατροπή : Αλγόριθμος \leftrightarrow ΔΡ

Αλγόριθμος Άσκηση

Διάβασε x, n

$m \leftarrow n$

$\zeta \leftarrow x$

αποτέλεσμα $\leftarrow 1$

Όσο $m > 0$ επανάλαβε

 Όσο $m \bmod 2 = 0$ επανάλαβε

$m \leftarrow m \text{ div } 2$

$z \leftarrow z * \zeta$

 Τέλος_επανάληψης


$m \leftarrow m - 1$

 αποτέλεσμα \leftarrow αποτέλεσμα * ζ

Τέλος_επανάληψης

Εκτύπωσε αποτέλεσμα

Τέλος Άσκηση



(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M7. Μετατροπή : Αλγόριθμος \leftrightarrow ΔΡ

Αλγόριθμος Άσκηση

$\beta \leftarrow 3$

$\tau \leftarrow 0$

Για i από 1 μέχρι 4

$\beta \leftarrow \beta + 2$

Για j από β μέχρι i με βήμα -2


$\tau \leftarrow \tau + j$

Τέλος επανάληψης

Τέλος επανάληψης

Εμφάνισε τ

Τέλος Άσκηση



(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M8. Επαναληπτικό μενού επιλογών με έξοδο

π.χ. Πρόγραμμα που με επαναληπτικό μενού επιλογών εκτελεί τις παρακάτω λειτουργίες:

1. Εμβαδό κύκλου 2. Εμβαδό τραπεζίου 3. Εμβαδό κυλίνδρου ($2\pi R^2 + 2\pi Rh$) 4. Έξοδος

Αρχή Επανάληψης

Γράψε "1. Εμβαδό κύκλου"

Γράψε "2. Εμβαδό τραπεζίου"

Γράψε "3. Εμβαδό κυλίνδρου"

Γράψε "4. Έξοδος"

Γράψε "Διάλεξε (1-4)"

Διάβασε επ

Αν (επ = 1) τότε

Διάβασε ρ

$E \leftarrow 3.14 * \rho^2$

Γράψε E

Αλλιώς Αν (επ = 2) τότε

Διάβασε β, Β, υ

$E \leftarrow (\beta + B) * \upsilon / 2$

Γράψε E

Αλλιώς Αν (επ = 3) τότε

Διάβασε R, h

$E \leftarrow 2 * 3.14 * R^2 + 2 * 3.14 * R * h$

Γράψε E

Αλλιώς Αν (επ <> 4) τότε

Γράψε "Λάθος επιλογή"

Τέλος Αν

Μέχρις Ότου (επ = 4)

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M9. Πίνακας τιμών μεταβλητών

Αλγόριθμος Άσκηση

$a \leftarrow 1$

Για i από 52 μέχρι 31

 με_βήμα-10

$\beta \leftarrow i - 11$

$\gamma \leftarrow 2 * \beta$

 Αν $a > 15$ τότε

$\gamma \leftarrow \gamma + a * \beta$

 Αλλιώς

$\gamma \leftarrow \gamma \text{ div } 3$

 Τέλος_αν

$a \leftarrow \beta - \gamma$

Τέλος_επανάληψης

Εκτύπωσε a, β, γ

Τέλος Άσκηση

	i	a	β	γ
		1		
1η επανάληψη	52		41	82
$1 > 15$, δεν ισχύει		14		27
2η επανάληψη	42		31	62
$14 > 15$, δεν ισχύει		11		20
3η επανάληψη	32		21	42
$11 > 15$, δεν ισχύει		7		14

Θα εκτυπωθούν οι τιμές 7, 21, 14.

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M9. Πίνακας τιμών μεταβλητών

Αλγόριθμος Άσκηση

$\beta \leftarrow 10$

Όσο $\beta \geq 0$ επανάλαβε

$a \leftarrow \beta + 3$

Αν $a < 8$ τότε

$\gamma \leftarrow a - \beta$

Αλλιώς

$\gamma \leftarrow a + \beta$

Τέλος_αν

Εκτύπωσε γ

$\beta \leftarrow \beta - 4$

Τέλος_επανάληψης

Τέλος_Άσκηση

	a	β	Υ
		10	
$10 \geq 0$, ισχύει - 1η επανάληψη	13	6	
$13 < 8$, δεν ισχύει			23
$6 \geq 0$, ισχύει - 2η επανάληψη	9	2	
$9 < 8$, δεν ισχύει			15
$2 \geq 0$, ισχύει - 3η επανάληψη	5	-2	
$5 < 8$, ισχύει			3
$-2 \geq 0$, δεν ισχύει			

Θα εκτυπωθούν οι τιμές του γ στην τελευταία στήλη: 23, 15, 3.

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M9. Πίνακας τιμών μεταβλητών

Αλγόριθμος Άσκηση

$x \leftarrow 1$

$a \leftarrow 2$

Αρχή_επανάληψης

$y \leftarrow x * a$

Αν $y \bmod 2 = 0$ τότε

$z \leftarrow x + 3$

$z \leftarrow x * z$

Αλλιώς

$x \leftarrow x + 2$

$z \leftarrow x + y$

Τέλος_αν

Εκτύπωσε z

$a \leftarrow a + 1$

Μέχρις_ότου $x > 7$

Εκτύπωσε x, y, z

Τέλος Άσκηση

Ο πίνακας τιμών των μεταβλητών έχει ως εξής:

	a	x	y	z
	2	1		
1η επανάληψη	3		2	
$2 \bmod 2 = 0$, ισχύει		4		8
$4 > 7$, δεν ισχύει - 2η επανάληψη	4		12	
$12 \bmod 2 = 0$, ισχύει		7		84
$7 > 7$, δεν ισχύει - 3η επανάληψη	5		28	
$28 \bmod 2 = 0$, ισχύει		10		280
$10 > 7$, ισχύει - τερματισμός επανάληψης				

Θα εκτυπωθούν οι τιμές 8, 84, 280, 10, 28, 280.

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M10. Εμφωλευμένες επαναλήψεις, για-για

π.χ. Πρόγραμμα που για καθένα από 25 αυτοκίνητα, διαβάζει: μοντέλο και 10 ποσοτικούς δείκτες. Εμφανίζει το μοντέλο με το μεγαλύτερο σύνολο δεικτών (χωρίς ισοτιμία)

```
max ← 0
για i από 1 μέχρι 25
  Διάβασε μ
  Σ ← 0
  για j από 1 μέχρι 10
    Διάβασε β
    Σ ← Σ + β
  ΤέλοςΕπανάληψης
  Αν (Σ > max) τότε
    max ← Σ
  maxM ← μ
ΤέλοςΑν
ΤέλοςΕπανάληψης
Γράψε maxM
```

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M10. Εμφωλευμένες επαναλήψεις, για-όσο/μέχρι

π.χ. Στους προκριματικούς αγώνες ακοντισμού συμμετέχουν 25 αθλητές. Πρόγραμμα που για κάθε αθλητή, διαβάζει το όνομά του και τις ρίψεις του, μέχρι να γίνουν 5 ή μέχρι μία ρίψη να ξεπεράσει το όριο πρόκρισης (75m). Εμφανίζει ποιοί και πόσοι προκρίνονται και ποιά η μέση επίδοση αυτών που προκρίνονται.

$\Sigma \leftarrow 0$

$\Pi \leftarrow 0$

για i από 1 μέχρι 25

Διάβασε on

$ρίψεις \leftarrow 0$

ΑρχήΕπανάληψης

Διάβασε ρ

$ρίψεις \leftarrow ρίψεις + 1$

ΜέχριςΌτου ($\rho > 75$ Η $ρίψεις = 5$)

Αν ($\rho > 75$) τότε

Γράψε "Προκρίνεται: ", on

$\Sigma \leftarrow \Sigma + \rho$

$\Pi \leftarrow \Pi + 1$

ΤέλοςΑν

ΤέλοςΕπανάληψης

Αν ($\Pi <> 0$) τότε

Γράψε "Προκρίνονται: ", Π , " αθλητές"

Γράψε "Μέση επίδοση: ", Σ/Π , " m"

Αλλιώς

Γράψε "Κανείς δεν προκρίθηκε"

ΤέλοςΑν

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M10. Εμφωλευμένες επαναλήψεις, όσο/μέχρι - για

π.χ. Πρόγραμμα που διαβάζει επαναληπτικά για κάθε μαθητή: όνομα και βαθμούς σε 10 μαθήματα, μέχρι να δοθεί για όνομα το κενό (""). Εμφανίζει για κάθε μαθητή το μέσο όρο του και τελικά το όνομα του μαθητή με τον μεγαλύτερο μέσο όρο (χωρίς ισοτιμία).

```
Π ← 0
Διάβασε ον
Όσο (ον <> "") επανάλαβε
  Π ← Π + 1
  Σ ← 0
  για i από 1 μέχρι 10
    Διάβασε β
    Σ ← Σ + β
  ΤέλοςΕπανάληψης
  μο ← Σ / 10
  Γράψε μο
  Αν (Π = 1) τότε
    max ← μο
    maxΟν ← ον
  Αλλιώς
    Αν (μο > max) τότε
      max ← μο
      maxΟν ← ον
  ΤέλοςΑν
ΤέλοςΑν
Διάβασε ον
ΤέλοςΕπανάληψης
Αν (Π <> 0) τότε
  Γράψε maxΟν
Αλλιώς
  Γράψε "Δεν δόθηκαν στοιχεία"
ΤέλοςΑν
```

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M10. Εμφωλευμένες επαναλήψεις, όσο/μέχρι - όσο/μέχρι

π.χ. Πρόγραμμα που διαβάζει επαναληπτικά για κάθε πωλητή: όνομα (μέχρι το κενό "") και πωλήσεις (μέχρι να δοθεί για πώληση το -1). Εμφανίζει για κάθε πωλητή το σύνολο των πωλήσεών του.

```
Διάβασε ον
Όσο (ον <> "") επανάλαβε
  Σ ← 0
  Διάβασε π
  Όσο (π <> -1) επανάλαβε
    Σ ← Σ + π
  Διάβασε π
ΤέλοςΕπανάληψης
Γράψε Σ
Διάβασε ον
ΤέλοςΕπανάληψης
```

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M11. Έλεγχος συνόλου γνωστού πλήθους εάν όλα τα στοιχεία του πληρούν μία ιδιότητα

π.χ. Πρόγραμμα που διαβάζει 100 θετικούς ακέραιους και ελέγχει εάν και οι 100 ήταν άρτιοι.

1 ^{ος} τρόπος	2 ^{ος} τρόπος	3 ^{ος} τρόπος (πιο αποδοτικός)
Π ← 0 για i από 1 μέχρι 100 Διάβασε α Αν (α mod 2 = 0) τότε Π ← Π + 1 ΤέλοςΑν ΤέλοςΕπανάληψης Αν (Π = 100) τότε Γράψε “ΝΑΙ” Αλλιώς Γράψε “ΟΧΙ” ΤέλοςΑν	άρτιοι ← Αληθής για i από 1 μέχρι 100 Διάβασε α Αν (α mod 2 <> 0) τότε άρτιοι ← Ψευδής ΤέλοςΑν ΤέλοςΕπανάληψης Αν (άρτιοι = Αληθής) τότε Γράψε “ΝΑΙ” Αλλιώς Γράψε “ΟΧΙ” ΤέλοςΑν	i ← 1 άρτιοι ← Αληθής Όσο (i <= 100 ΚΑΙ άρτιοι = Αληθής) επανάλαβε Διάβασε α Αν (α mod 2 <> 0) τότε άρτιοι ← Ψευδής i ← i + 1 ΤέλοςΕπανάληψης Αν (άρτιοι = Αληθής) τότε Γράψε “ΝΑΙ” Αλλιώς Γράψε “ΟΧΙ” ΤέλοςΑν

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M12. Εξάντληση αποθέματος

Ένας μανάβης προμηθεύεται 300 κιλά μήλα για να τα πουλήσει με 2€ το κιλό. Να γραφεί πρόγραμμα το οποίο για κάθε παραγγελία διαβάζει την ποσότητα σε κιλά και εφόσον το απόθεμα επαρκεί για την κάλυψη της παραγγελίας να την εκτελεί. Διαφορετικά να διαθέτει το διαθέσιμο απόθεμα. Η εισαγωγή τερματίζεται όταν εξαντληθεί το απόθεμα. Για κάθε παραγγελία να εμφανίζει το κόστος της.

```
ΑΠ ← 300
Όσο (ΑΠ<>0) επανάλαβε
  Διάβασε ποσ
  Αν (ποσ > ΑΠ) τότε
    ποσ ← ΑΠ
  ΤέλοςΑν
  ΑΠ ← ΑΠ – ποσ
  Γράψε ποσ*2, " €"
ΤέλοςΕπανάληψης
```

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M13. Αποφυγή παραβίασης άνω / κάτω φράγματος

π.χ. Ένα ασανσέρ έχει μέγιστο όριο φορτίου τα 350 κιλά. Πρόγραμμα που δέχεται επαναληπτικά το βάρος του κάθε ατόμου που εισέρχεται σε αυτό μέχρι να παραβιασθεί αυτό το όριο. Εμφανίζει τελικά το πλήθος των ατόμων που επιτρέπεται να εισέλθουν καθώς και το συνολικό τους βάρος.

1ος τρόπος	2ος τρόπος
$\Pi \leftarrow 0$	$\Pi \leftarrow 0$
$SB \leftarrow 0$	$Υπ \leftarrow 350$! υπόλοιπο
Διάβασε β	Διάβασε β
Όσο ($SB + \beta \leq 350$) επανάλαβε	Όσο ($Υπ \geq \beta$) επανάλαβε
$SB \leftarrow SB + \beta$	$Υπ \leftarrow Υπ - \beta$
$\Pi \leftarrow \Pi + 1$	$\Pi \leftarrow \Pi + 1$
Διάβασε β	Διάβασε β
Τέλος Επανάληψης	Τέλος Επανάληψης
Γράψε Π, SB	$SB \leftarrow 350 - Υπ$
	Γράψε Π, SB

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M14. Ανάλυση ακέραιου στα ψηφία του

π.χ. Εισαγωγή θετικού ακέραιου και εμφάνιση του συμμετρικού του: του αριθμού που προκύπτει αν διαβαστεί ανάποδα.

ΠΡΟΓΡΑΜΜΑ symmetrikos
ΜΕΤΑΒΛΗΤΕΣ

ΑΚΕΡΑΙΕΣ: x, sym

ΑΡΧΗ

ΓΡΑΨΕ 'Δώσε θετικό ακέραιο'

ΔΙΑΒΑΣΕ x

sym <-- 0

ΟΣΟ x <> 0 **ΕΠΑΝΑΛΑΒΕ**

sym <-- sym*10 + x **mod** 10

x <-- x **div** 10

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΓΡΑΨΕ 'Ο συμμετρικός του είναι ο: ', sym

ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ

```
Δώσε θετικό ακέραιο
34876129
Ο συμμετρικός του είναι ο: 92167843
```

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M15. Γνωστό μέγιστο πλήθος επαναλήψεων (Όσο/Μέχρι)

π.χ. Σε ένα τηλεπαιχνίδι ο παίκτης ξεκινάει με 300000€ και καλείται να απαντήσει σε 7 ερωτήσεις πολλαπλής επιλογής. Σε κάθε ερώτηση ο παίκτης μπορεί να ποντάρει τα χρήματα που διαθέτει στις προτεινόμενες απαντήσεις. Προχωράει στην επόμενη ερώτηση με τα χρήματα που πόνταρε στη σωστή απάντηση. Σταματάει το παιχνίδι εάν μείνει χωρίς χρήματα ή εφόσον τελειώσει και την έβδομη ερώτηση. Πρόγραμμα που προσομοιώνει το παιχνίδι ως εξής: σε κάθε ερώτηση και εφόσον ο παίκτης έχει ακόμη χρήματα, να διαβάσει το ποσό που πόνταρε στη σωστή απάντηση με έλεγχο εγκυρότητας ώστε να είναι το πολύ όσο το και το ποσό που του έχει μείνει. Τελικά να εμφανίζει τα χρήματα που κέρδισε ή την ερώτηση (1-7) στην οποία τα έχασε όλα.

$x \leftarrow 300000$ $ερ \leftarrow 0$

Όσο ($ερ < 7$ ΚΑΙ $x <> 0$) επανάλαβε

$ερ \leftarrow ερ + 1$

ΑρχήΕπανάληψης

Διάβασε ποσό

ΜέχριςΌτου ($ποσό \geq 0$ ΚΑΙ $ποσό \leq x$)

$x \leftarrow ποσό$

ΤέλοςΕπανάληψης

Αν ($x <> 0$) τότε

Γράψε 'Κέρδισε ', x , '€'


Αλλιώς

Γράψε 'Έχασε στην ερώτηση: ', $ερ$

ΤέλοςΑν

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M16. Καταγραφή των στοιχείων ενός συνόλου που πληρούν μία ιδιότητα



όλα τα στοιχεία $\Leftrightarrow \Pi = N$

κάποια από τα στοιχεία $\Leftrightarrow \Pi \neq 0$

κανένα από τα στοιχεία $\Leftrightarrow \Pi = 0$

% ποσοστό στοιχείων $\Leftrightarrow \Pi/N*100$

maxΣερι $\leftarrow 0$

σερι $\leftarrow 0$

Για κάθε στοιχείο x του συνόλου

Αν (το x πληροί την ιδιότητα) τότε

σερι \leftarrow σερι + 1

Αλλιώς

σερι $\leftarrow 0$

ΤέλοςΑν

Αν (σερι > maxΣερι) τότε

maxΣερι \leftarrow σερι

ΤέλοςΑν

ΤέλοςΕπανάληψης

Γράψε maxΣερι

μέγιστο σερί στοιχείων

(§ 2.4.5 - § 8.2) Δομή Επανάληψης – Μεθοδολογίες Ασκήσεων

M17. Παιχνίδι με 2 παίκτες

α) τελειώνει στους N γύρους

νικές1 \leftarrow 0

νικές2 \leftarrow 0

για γύρο από 1 μέχρι N

! βγάλε νικητή γύρου ...

Αν (νικητής ο 1^{ος}) τότε

νικές1 \leftarrow νικές1 + 1

Αλλιώς Αν (νικητής ο 2^{ος}) τότε

νικές2 \leftarrow νικές2 + 1

Τέλος Αν

Τέλος Επανάληψης

β) τελειώνει στις N νίκες

νικές1 \leftarrow 0

νικές2 \leftarrow 0

Όσο (νικές1 < N ΚΑΙ νικές2 < N) επανάλαβε

! βγάλε νικητή γύρου ...

Αν (νικητής ο 1ος) τότε

νικές1 \leftarrow νικές1 + 1

Αλλιώς Αν (νικητής ο 2ος) τότε

νικές2 \leftarrow νικές2 + 1

Τέλος Αν

Τέλος Επανάληψης

- νικές1 > νικές2 \Rightarrow τελικός νικητής ο 1^{ος}
- νικές2 > νικές1 \Rightarrow τελικός νικητής ο 2^{ος}
- νικές1 = νικές2 \Rightarrow Ισοπαλία


ή ... Μέχρις Ότου (νικές1 = N Η νικές2 = N)

- νικές1 = N \Rightarrow τελικός νικητής ο 1^{ος}
- νικές2 = N \Rightarrow τελικός νικητής ο 2^{ος}

Ανασκόπηση

	Επαναληπτικά προβλήματα	
Δομή Επανάληψης	Γνωστού πλήθους επαναλήψεων	Αγνώστου πλήθους επαναλήψεων
Όσο	✓	✓
Μέχρι	✓	✓
Για	✓ <i>(προτιμότερη)</i>	✗

(§ 8.2.3) Κανόνες χρήσης των εμφωλευμένων βρόχων



Βρόχος: τμήμα εντολών που εσωκλείεται σε μία δομή επανάληψης

Κανόνες χρήσης των εμφωλευμένων βρόχων:

1. Ο εσωτερικός βρόχος πρέπει να βρίσκεται ολόκληρος μέσα στον εξωτερικό. Ο βρόχος που ξεκινάει τελευταίος, πρέπει να ολοκληρώνεται πρώτος.
2. Η είσοδος σε κάθε βρόχο υποχρεωτικά γίνεται από την αρχή του.
3. Δεν μπορεί να χρησιμοποιηθεί η ίδια μεταβλητή ως μετρητής δύο ή περισσοτέρων βρόχων που ο ένας βρίσκεται στο εσωτερικό του άλλου

(§ 2.4.5) Πολλαπλασιασμός αλά ρωσικά

$\gamma \leftarrow 0$
Διάβασε $a, \beta \neq 0$
Όσο ($\beta > 0$) επανάλαβε
 Αν ($\beta \bmod 2 \neq 0$) τότε
 $\gamma \leftarrow \gamma + a$
 ΤέλοςΑν
 $a \leftarrow a * 2$
 $\beta \leftarrow \beta \text{ div } 2$
ΤέλοςΕπανάληψης
Γράψε γ

45	19	45
90	9	90
180	4	
360	2	
720	1	720
Άθροισμα = 855		

Ολίσθηση (shift)

Στα κυκλώματα του υπολογιστή τα δεδομένα αποθηκεύονται με δυαδική μορφή, δηλαδή 0 και 1, ανεξάρτητα από το πως τα ορίζει ο προγραμματιστής, όπως ακεραίους ή πραγματικούς σε δεκαδικό σύστημα, ή ακόμη χαρακτήρες κ.λπ. Έτσι ο αριθμός 17 του δεκαδικού συστήματος ισοδυναμεί με τον αριθμό 00010001 του δυαδικού συστήματος, ο οποίος μπορεί να αποθηκευθεί σε ένα byte. Αν μετακινήσουμε τα ψηφία αυτά κατά μία θέση προς τα αριστερά, δηλαδή αν προσθέσουμε ένα 0 στο τέλος του αριθμού και αγνοήσουμε το αρχικό 0, τότε προκύπτει ο αριθμός 00100010 του δυαδικού συστήματος, που ισοδυναμεί με το αριθμό 34 του δεκαδικού συστήματος. Επίσης, με παρόμοιο τρόπο, αν μετακινήσουμε τα ψηφία κατά μία θέση δεξιά, δηλαδή αποκόψουμε το τελευταίο ψηφίο 1 και θεωρήσουμε ένα ακόμη αρχικό 0, τότε προκύπτει ο αριθμός 00001000 του δυαδικού συστήματος, που ισοδυναμεί με τον αριθμό 8 του δεκαδικού συστήματος. Άρα η ολίσθηση προς τα αριστερά ισοδυναμεί με πολλαπλασιασμό επί δύο, ενώ η ολίσθηση προς τα δεξιά ισοδυναμεί με την ακέραια διαίρεση διά δύο.

Τέλος Κεφ. 2,7,8